

Е. А. СИДОРОВА, С. П. ЖЕЛЕЗНЯК

**ЛАБОРАТОРНЫЙ ПРАКТИКУМ ПО РАБОТЕ
В ТАБЛИЧНОМ ПРОЦЕССОРЕ MICROSOFT EXCEL 2010**

ЧАСТЬ 2

СОЗДАНИЕ ФОРМУЛ. ВСТРОЕННЫЕ ФУНКЦИИ

ОМСК 2018

Министерство транспорта Российской Федерации
Федеральное агентство железнодорожного транспорта
Омский государственный университет путей сообщения

Е. А. Сидорова, С. П. Железняк

ЛАБОРАТОРНЫЙ ПРАКТИКУМ ПО РАБОТЕ
В ТАБЛИЧНОМ ПРОЦЕССОРЕ MICROSOFT EXCEL 2010

Часть 2

СОЗДАНИЕ ФОРМУЛ. ВСТРОЕННЫЕ ФУНКЦИИ

Утверждено методическим советом университета
в качестве практикума к выполнению самостоятельной и
лабораторных работ

Омск 2018

УДК 004.67(075.8)
ББК 32.973я73
С34

Лабораторный практикум по работе в табличном процессоре Microsoft Excel 2010. Часть 2. Создание формул. Встроенные функции /
Е. А. Сидорова, С. П. Железняк; Омский гос. ун-т путей сообщения. Омск, 2018. 37 с.

Лабораторный практикум состоит из трех частей. Во второй части рассмотрены особенности создания формул и применения различных встроенных функций в табличном процессоре Microsoft Excel 2010. Представлены примеры организации вычислений и практические задания.

Практикум предназначен для студентов и аспирантов очной и заочной форм обучения всех направлений подготовки (специальностей), изучающих дисциплины информационного профиля: «Информатика», «Информационные технологии», «Компьютерные технологии и информатика», «Информационные системы и базы данных» и др., а также может быть использован для самостоятельной работы любых категорий пользователей персонального компьютера.

Библиогр.: 5 назв. Табл. 3. Рис. 25.

Рецензенты: доктор техн. наук, профессор В. Н. Горюнов;
доктор техн. наук, профессор А. А. Кузнецов.

ОГЛАВЛЕНИЕ

Введение	5
Лабораторная работа 2. Создание формул. Встроенные функции	6
2.1. Ввод формул	6
2.2. Отображение и редактирование формул	7
2.3. Встроенные функции	9
2.4. Мастер функций	14
2.5. Кнопка Автосумма	18
2.6. Отображение результатов вычислений в строке состояния	19
2.7. Абсолютные и относительные ссылки.....	20
2.8. Работа с массивами данных	25
2.9. Функция СУММПРОИЗВ	28
2.10. Контрольные вопросы	30
2.11. Задания	31
Задание 1. Ввод формул. Абсолютные и относительные ссылки.	31
Задание 2. Смешанные ссылки.	32
Задание 3. Применение встроенных функций КОРЕНЬ и LOG.....	33
Задание 4. Работа с массивом данных.....	33
Задание 5. Преобразования чисел в разных системах счисления.	34
Задание 6. Решение задач по теме «Измерение информации».....	35
Библиографический список.....	36

ВВЕДЕНИЕ

Табличный процессор Microsoft Excel 2010 (далее – Excel) является одним из самых мощных и распространенных в мире приложений для представления, эффективной обработки и анализа числовой информации. С этой целью в нем реализован широкий спектр возможностей для осуществления вычислений с помощью формул и множества встроенных функций (математических, инженерных, финансовых, статистических и др.). Преимущество электронных таблиц Excel заключается в автоматизации процесса выполнения расчетов и занесения результатов в графы таблиц. При этом изменение исходных данных автоматически ведет к мгновенному пересчету результатов.

В лабораторной работе 2 рассматриваются режимы создания и ввода формул в ячейки электронных таблиц, приведены особенности использования встроенных функций, представлены задания для самостоятельного выполнения студентом. После завершения лабораторной работы студент предъявляет преподавателю файл с выполненным заданием.

Библиографический список, представленный в конце практикума, содержит литературу для углубленного изучения материала по рассматриваемой тематике.

СОЗДАНИЕ ФОРМУЛ. ВСТРОЕННЫЕ ФУНКЦИИ

Цель работы:

- 1) изучение особенностей создания формул в табличном процессоре Excel;
- 2) освоение способов ввода и редактирования формул и применения в них встроенных функций;
- 3) закрепление навыков создания формул для автоматизации вычислений в Excel.

2.1. Ввод формул

Ввод в ячейку Excel информации, которая определяется расчетным путем, должен осуществляться не в виде готового результата (вычисленного, например, с помощью калькулятора), а в виде **формулы** – математического выражения, которое создается для вычисления результата и может зависеть от содержимого других ячеек. Только в этом случае изменение данных в исходных ячейках автоматически повлечет за собой пересчет по формуле и отображение нового значения в результирующей ячейке.

Любая формула в ячейке Excel начинается со знака равенства (=). По этому отличительному признаку табличный процессор автоматически определяет в ячейке тип информации (расчетную формулу), требующий от него вычислительных действий и отображения результата. Формулы могут содержать адреса или имена ячеек с данными, константы, знаки арифметических действий, операции сравнения, функции с аргументами и другие символы. Для изменения порядка выполнения вычислений в формуле применяют круглые скобки.

Формулы в ячейках Excel можно создавать разными способами: набирать вручную на клавиатуре, а также применять для этого широкий арсенал средств автоматизированного ввода и редактирования информации. Рассмотрим основные приемы создания простейших формул в Excel на конкретном примере.

Пример 1. Для исходных чисел в ячейках A2:B3 (рис. 1) вычислить:

- 1) в ячейке C2 – сумму значения ячейки A2 и умноженного на 4 значения ячейки B2;
- 2) в ячейке C3 – квадрат разности значений ячеек A3 и B3;
- 3) в ячейке C4 – отношение полученных в ячейках C2 и C3 результатов.

Порядок действий:

1) установить курсор в ячейку C2, где требуется получить результат первого расчета;

2) ввести знак «=» (при этом он автоматически появится и в строке формул, где далее последовательно будут отображаться все вводимые в ячейку символы формулы);

3) набрать текст формулы, указывая адреса ячеек (их удобно вводить, щелкая левой кнопкой мыши на нужных ячейках рабочего листа), знаки арифметических операций и числовую константу 4: =A2+B2*4;

4) нажать *Enter*. В ячейке C2 будет отображаться результат вычислений – число 31, а в строке формул – полный текст формулы (см. рис. 1).

Аналогичным образом вводятся формулы в ячейки C3 и C4 (см. рис. 1).

Теперь, если изменить любое число в ячейках A2:B3, результат в ячейках C2:C4 будет пересчитан автоматически.

	A	B	C	
1	Исходные данные		Результат	Формула в столбце "Результат"
2	3	7	31	=A2+B2*4
3	5	11	36	=(B3-A3)^2
4	-	-	0,86	=C2/C3

Рис. 1. Ввод формул в ячейки Excel

Как и любое содержимое ячейки, всю формулу или отдельные ее элементы можно набирать не непосредственно в ячейке, а в строке формул.

Если расчет по формуле не получился, то зачастую причина ошибки заключается в том, что в начале формулы отсутствует знак «=». Коды возможных ошибок при использовании формул Excel подробно описаны в работе [2, с. 34 – 36].

2.2. Отображение и редактирование формул

Редактирование ячейки с формулой выполняется так же, как и любой другой ячейки – либо непосредственно в поле ячейки (при этом ячейки, входящие в формулу, автоматически выделяются разными цветами соответственно

ссылкам на них), либо в строке формул. После изменения формулы следует нажать *Enter* или *Tab*.

Включить (или, наоборот, отключить) режим одновременного просмотра всех введенных на листе Excel формул можно с помощью команды **Формулы** → [Зависимости формул] → *Показать формулы* (рис. 2, а) или установки соответствующего флажка в меню **Файл** → Параметры → Дополнительно (рис. 2, б).




Рис. 2. Установка режима отображения формул в меню **Формулы** (а) и в меню **Файл** (б)

В некоторых случаях, особенно при поиске причин ошибок в расчетах, целесообразно отобразить связи между ячейками и формулами. При этом ячейки, на которые ссылаются формулы, называются *влияющими* ячейками, а ячейки, которые содержат формулы со ссылками на другие ячейки, называются *зависимыми* ячейками. Для отображения связей нужно установить курсор в ячейку и выполнить команду **Формулы** → [Зависимости формул] → *Влияющие ячейки* и (или) **Формулы** → [Зависимости формул] → *Зависимые ячейки* (см. рис. 2, а). Каждую связь Excel отображает синей линией, начало которой обозначается точкой и находится во влияющей ячейке, а конец обозначается стрелкой и находится в зависимой ячейке (рис. 3). Такую операцию последовательно можно проделать для всех анализируемых ячеек. Убрать все стрелки или только стрелки нужного типа можно с помощью команды **Формулы** → [Зависимости формул] → *Убрать стрелки* (см. рис. 2, а).

	А	В	С
	Исходные данные		Результат
1			
2	3	7	31
3	5	11	36
4	-	-	0,86

Рис. 3. Отображение связей влияющих и зависимых ячеек

2.3. Встроенные функции

В подавляющем большинстве случаев при решении научных и производственных задач недостаточно выполнять только простые арифметические действия, а необходимо вычислять сложные математические выражения, содержащие различные функции. С этой целью в Excel имеется более 400 встроенных стандартных функций, которые систематизированы по 11 категориям в зависимости от их функционального назначения (математические, статистические, логические, инженерные и др.).

Запись любой функции Excel начинается с имени функции, сразу после которого в круглых скобках записываются ее *аргументы* – исходные данные для расчета значения функции. Имена функций обозначаются прописными буквами, но при их вводе Excel нечувствителен к регистру символов, поэтому имена можно вводить любыми буквами – и прописными, и строчными.

Разные функции могут иметь как строго фиксированное, так и неопределенное количество аргументов. Для многих функций важен порядок следования аргументов. Если функция имеет более одного аргумента, то они разделяются точкой с запятой.

Общий формат записи встроенной функции Excel имеет вид:

ИМЯ_ФУНКЦИИ(Аргумент1; Аргумент2; ...; АргументN).

В качестве аргументов функции могут выступать числовые значения, математические выражения и операторы, ссылки на ячейки (диапазон ячеек) или другие функции. Если в качестве аргументов функции используются другие функции, то такие функции называются *вложенными*. Некоторые функции не имеют аргументов и записываются с пустыми скобками, например, функция СЕГОДНЯ(), возвращающая текущую дату.

Для вставки функции в формулу ее можно набрать вручную или применить специальные средства автоматизированного ввода – *Мастер функций*, кнопку *Автосумма*, которые будут рассмотрены ниже.

При ручном вводе имени функции Excel предлагает всплывающий список похожих имен функций, из которого двойным щелчком мыши можно выбрать нужную функцию и вставить ее в ячейку, после чего продолжить ввод аргументов. При этом под набираемой функцией появляется всплывающая подсказка по ее синтаксису.

Рассмотрим ручной ввод функции на примере самой популярной функции Excel СУММ, применяемой для суммирования значений в ячейках. Для вычисления с помощью этой функции в примере 1 результата в ячейке С2 (см. рис. 1) необходимо выполнить следующие действия:

- 1) установить указатель мыши в ячейку С2 и ввести знак «=»;
- 2) ввести первые символы имени функции (достаточно символов «су») и двойным щелчком выбрать из появившегося списка похожих имен функцию СУММ (рис. 4, а) – в результате в ячейке появится имя СУММ и открывающая скобка (вместо этого можно было полностью набрать на клавиатуре слово СУММ и открывающую скобку);
- 3) для ввода первого аргумента функции щелкнуть левой кнопкой мыши на ячейке А2 (адрес ячейки будет добавлен в формулу);
- 4) ввести знак точки с запятой «;» – разделителя аргументов функции;
- 5) для ввода второго аргумента функции щелкнуть левой кнопкой мыши на ячейке В2, набрать на клавиатуре знак умножения «*» и число 4;
- 6) поставить закрывающую скобку (рис. 4, б);
- 7) нажать *Enter*. В результате текст формулы =СУММ(А2; В2*4) будет отображаться в строке формул, а в ячейке С2 будет показан результат 31.




Рис. 4. Ручной ввод формулы с функцией СУММ в ячейку Excel: всплывающий список имен функций (а) и окончательный вид формулы (б)

Если вычисления выполняются для диапазона смежных ячеек, то в качестве аргументов функции не перечисляют его отдельные ячейки, а записывают ссылку на весь диапазон с указанием первой и последней ячеек, разделяемых двоеточием. Например, для вычисления суммы значений диапазона ячеек с А2 по В3 формула будет иметь вид: =СУММ(А2:В3).

Наиболее востребованные на практике встроенные функции Excel из разных категорий приведены в табл. 1, примеры их использования – на рис. 5, а – д. В записи формата функции в квадратные скобки заключены необязательные аргументы. В примерах в качестве аргументов функций указаны ссылки на адреса ячеек Excel.

Функции для работы с массивами приведены в подразд. 2.8. Функции для обработки данных, удовлетворяющих заданным условиям, и логические функции рассмотрены в третьей части практикума.

Таблица 1

Встроенные функции Excel

Формат функции	Результат
1	2
Категория: <i>Дата и время</i>	
СЕГОДНЯ()	Текущая дата
ТДАТА()	Текущие дата и время
Категория: <i>Математические</i>	
СУММ(Число1; [Число2]; ...)	Сумма чисел
ОКРУГЛ(Число; Число_разрядов)	Округление числа до указанного количества десятичных разрядов (остальные разряды отбрасываются и в дальнейших расчетах не учитываются)
ABS(Число)	Модуль (абсолютная величина) числа
EXP(Число)	Экспонента заданного числа
LN(Число)	Натуральный логарифм числа
LOG(Число; [Основание])	Логарифм числа по заданному основанию (если основание не задано, то по основанию 10)
LOG10(Число)	Десятичный логарифм числа
COS(Число)	Косинус угла (Число – угол в радианах)
SIN(Число)	Синус угла (Число – угол в радианах)
TAN(Число)	Тангенс угла (Число – угол в радианах)
КОРЕНЬ(Число)	Значение квадратного корня числа
СТЕПЕНЬ(Число; Степень)	Результат возведения числа в степень
ПИ()	Округленное до 15 знаков после запятой число Пи

1	2
Категория: <i>Статистические</i>	
МАКС(Число1; [Число2]; ...)	Наибольшее из чисел
МИН(Число1; [Число2]; ...)	Наименьшее из чисел
СРЗНАЧ(Число1; [Число2]; ...)	Среднее арифметическое чисел
СЧЁТ(Значение1; [Значение2]; ...)	Количество числовых значений в диапазоне ячеек
Категория: <i>Инженерные</i> (представлены функции преобразования чисел из одной системы счисления в другую, разрядность – количество знаков в записи числа, указывается при необходимости приписать возвращаемому значению ведущие нули)	
ДЕС.В.ВОСЬМ(Число; [Разрядность])	Десятичное число в восьмеричное
ДЕС.В.ДВ(Число; [Разрядность])	Десятичное число в двоичное
ДЕС.В.ШЕСТН(Число; [Разрядность])	Десятичное число в шестнадцатеричное
ДВ.В.ВОСЬМ(Число; [Разрядность])	Двоичное число в восьмеричное
ДВ.В.ДЕС(Число)	Двоичное число в десятичное
ДВ.В.ШЕСТН(Число; [Разрядность])	Двоичное число в шестнадцатеричное
ВОСЬМ.В.ДВ(Число; [Разрядность])	Восьмеричное число в двоичное
ВОСЬМ.В.ДЕС(Число)	Восьмеричное число в десятичное
ВОСЬМ.В.ШЕСТН(Число; [Разрядность])	Восьмеричное число в шестнадцатеричное
ШЕСТН.В.ВОСЬМ(Число; [Разрядность])	Шестнадцатеричное число в восьмеричное
ШЕСТН.В.ДВ(Число; [Разрядность])	Шестнадцатеричное число в двоичное
ШЕСТН.В.ДЕС(Число)	Шестнадцатеричное число в десятичное
Категория: <i>Текстовые</i>	
СЦЕПИТЬ(Текст1; [Текст2]; ...)	Объединение нескольких текстовых строк в одну
ДЛСТР(Текст)	Определение количества символов в текстовой строке

Формат данных *Дата* в ячейке позволяет осуществлять некоторые математические действия с датами. Например, количество дней между датами определяется как разность дат, а для определения даты, отстоящей от начальной даты на заданное количество дней вперед (или назад), нужно прибавить (или вычесть) это количество дней к начальной дате (см. рис. 5, а).

Для вычисления корня n -й степени помимо встроенной функции СТЕПЕНЬ можно применять известную математическую формулу

$$\sqrt[n]{x^k} = x^{k/n}.$$

	А	В	
1	Категория: <i>Дата и время</i>		
2	Данные	Результат	Формула в столбце "Результат"
3		28.03.2018	=СЕГОДНЯ()
4		28.03.2018 16:15	=ТДАТА()
5	12.08.1999	6803	=В3-А5
6		11.10.2018	=А5+7000
7	В ячейке В5 вычислено <i>количество</i> дней между датами в ячейках В3 и А5.		
8	В ячейке В6 вычислена <i>дата</i> , отстоящая на 7000 дней вперед относительно даты в ячейке А5		

а

	А	В	С	Д	
1	Категория: <i>Математические</i>				
2	Данные		Результат	Формула в столбце "Результат"	
3	12,37	55	11,08	78,45	=СУММ(А3:С3)
4				23,45	=СУММ(А3;С3)
5				23,5	=ОКРУГЛ(Д4;1)
6	-8			8	=ABS(А6)
7	3			20,0855	=EXP(А7)
8	49			7	=КОРЕНЬ(А8)
9	25			2,9240	=СТЕПЕНЬ(А9;1/3)
10				2,9240	=А9^(1/3)
11	7			1,9459	=LN(А11)
12	7	2		2,8074	=LOG(А12;В12)
13	7			0,8451	=LOG10(А13)
14	4,5			-0,2108	=COS(А14)
15	4,5			-0,9775	=SIN(А15)
16	4,5			4,6373	=TAN(А16)

б

	А	В	
1	Категория: <i>Инженерные</i>		
2	Данные	Результат	Формула в столбце "Результат"
3	47	57	=ДЕС.В.ВОСЬМ(А3)
4	47	101111	=ДЕС.В.ДВ(А4)
5	47	2F	=ДЕС.В.ШЕСТН(А5)
6			
7	101111	57	=ДВ.В.ВОСЬМ(А7)
8	101111	47	=ДВ.В.ДЕС(А8)
9	101111	2F	=ДВ.В.ШЕСТН(А9)
10			
11	57	101111	=ВОСЬМ.В.ДВ(А11)
12	57	47	=ВОСЬМ.В.ДЕС(А12)
13	57	2F	=ВОСЬМ.В.ШЕСТН(А13)
14			
15	2F	57	=ШЕСТН.В.ВОСЬМ(А15)
16	2F	101111	=ШЕСТН.В.ДВ(А16)
17	2F	47	=ШЕСТН.В.ДЕС(А17)

в

	А	В	С	Д	
1	Категория: <i>Статистические</i>				
2	23	18	22	23	=МАКС(А2:С2)
3	23	18	22	18	=МИН(А3:С3)
4	7	3	9	6,33333	=СРЗНАЧ(А4:С4)
5	8		11	2	=СЧЁТ(А5:С5)

г

	А	В	С	Д	
1	Категория: <i>Текстовые</i>				
2	Данные		Результат	Формула в столбце "Результат"	
3	Зуев	И.	К.	Зуев И.К.	=СЦЕПИТЬ(А3; " "; В3; С3)
4				4	=ДЛСТР(А3)
5				9	=ДЛСТР(Д3)

д

Рис. 5. Примеры применения встроенных функций Excel: дата и время (а), математические (б), инженерные (в), статистические (г), текстовые (д)

При записи формул в Excel следует различать операции, которые относятся непосредственно к функции или к ее аргументу:

операция, которая относится к функции, записывается после скобки, закрывающей список аргументов функции;

операция, которая относится к аргументу функции, записывается непосредственно после него внутри скобок, ограничивающих список аргументов функции.

В одной формуле могут быть как одна, так и несколько функций, объединенных между собой различными знаками математических операций. Иногда бывает удобно присвоить ячейкам Excel смысловые имена и использовать их в формуле вместо адресов [4, п. 1.3.1]. Такой подход делает формулу более наглядной, но усложняет визуальный контроль входящих в нее значений и поиск их на листе Excel. В табл. 2 приведены примеры записи математических выражений в Excel (при этом имеется в виду, что значение x хранится в ячейке A1, y – в ячейке A2 и указанным ячейкам присвоены соответствующие имена x и y).

Таблица 2

Примеры записи математических выражений в Excel



Математическая запись	Запись в Excel	
	с именами ячеек	с адресами ячеек
$\sin^3 x$	SIN(x)^3	SIN(A1)^3
$\sin x^2$	SIN(x^2)	SIN(A1^2)
$\sin^3 x^2$	SIN(x^2)^3	SIN(A1^2)^3
$\sin x-5 $	SIN(ABS(x-5))	SIN(ABS(A1-5))
$2x^5 - e^{x+y}$	2*x^5 - EXP(x+y)	2*A1^5 - EXP(A1+A2)
$\cos 7x - \sqrt[3]{y^4}$	COS(7*x) - y^(4/3)	COS(7*A1) - A2^(4/3)
$\sqrt{\operatorname{tg}^2 x + 6}$	КОРЕНЬ(TAN(x)^2+6)	КОРЕНЬ(TAN(A1)^2+6)
$\frac{5}{xy} + 8y$	5 / (x*y) + 8*y или 5 / x / y + 8*y	5 / (A1*A2) + 8*A2 или 5 / A1 / A2 + 8*A2
$\frac{\lg x + \ln y}{2x - 3}$	(LOG10(x) + LN(y)) / (2*x - 3)	(LOG10(A1) + LN(A2)) / (2*A1 - 3)

2.4. Мастер функций

Далеко не все функции Excel имеют такое простое определение, как, например, функция СУММ. Некоторые из них имеют сложное синтаксическое

написание, а также требуют много аргументов различных типов. Набор таких функций вручную может представлять определенные трудности. С целью упрощения ввода сложных функций и уменьшения количества ошибок в Excel существует специальная программа *Мастер функций*, позволяющая вводить функции в полуавтоматическом режиме. *Мастер функций* содержит два шага – выбор функции и настройку ее аргументов.

Запуск *Мастера функций* можно осуществить несколькими способами (рис. 6):

- 1) щелкнуть кнопку  слева от строки формул (преимуществом этого способа является то, что он доступен из любого места рабочего листа);
- 2) выполнить команду **Формулы** → [Библиотека функций] →  *Вставить функцию*;
- 3) установить курсор в ячейку результата, ввести знак «=», затем из списка функций в поле адреса ячейки выбрать опцию *Другие функции*.




Рис. 6. Способы запуска *Мастера функций*

Рассмотрим применение *Мастера функций* на примере ввода функции СУММ в примере 1 (см. рис. 1). Для этого нужно выполнить следующие действия.

1. Любым способом запустить *Мастер функций* (см. рис. 6).
2. В открывшемся диалоговом окне *Мастер функций* – шаг 1 из 2 (рис. 7) выбрать в разделе *Категория* нужную категорию функций – *Математические* (если категория неизвестна, можно выбрать опцию *Полный алфавитный перечень*), в разделе *Выберите функцию* – нужную функцию СУММ. При выделении функции в нижней части окна *Мастера функций* появляется ее краткое описание (см. рис. 7).

3. После выбора функции ее имя автоматически заносится в строку формул вместе со скобками, ограничивающими список аргументов, и одновременно открывается окно *Аргументы функции* (рис. 8). В левом верхнем углу этого окна указывается имя выбранной функции, а ниже – поля для ввода каждого аргумента. Справа от поля аргумента выводится его текущее значение. Внизу окна отображается справочная информация, указывающая назначение функции и вводимого аргумента, а также текущий результат вычисления.




Рис. 7. Окно выбора функции






Рис. 8. Окно ввода аргументов функции

В нашем примере в поле *Число1* нужно ввести адрес ячейки, где находится первое слагаемое (A2). Для этого можно набрать его вручную на клавиатуре или щелкнуть левой кнопкой мыши на ячейке A2.

4. Как и любое другое окно, диалоговое окно *Мастера функций* при необходимости можно передвинуть по экрану, однако оно все равно частично или полностью может закрывать нужные ячейки. В таком случае при указании в качестве аргумента функции ссылок на ячейки удобно временно уменьшить окно *Мастера функций*, нажав кнопку  в правой части поля ввода соответствующего аргумента (см. рис. 8). При этом диалоговое окно *Мастера функций* сворачивается до размера строки (рис. 9), открывая доступ к ячейкам. После того, как требуемые ячейки будут выделены и введены необходимые элементы формулы, нужно снова развернуть окно *Мастера функций* до полного размера, нажав в его правой части кнопку  (см. рис. 9). В нашем примере мы проделали это для ввода в поле *Число2* аргумента B2*4 (второго слагаемого суммы) (см. рис. 8 и 9).

Следует отметить, что обычно диалоговое окно *Мастера функций* уменьшают до размера строки лишь в тех случаях, когда оно закрывает доступ к нужным ячейкам.

5. После ввода всех аргументов функции в окне *Мастера функций* нужно щелкнуть кнопку *OK* (рис. 11), и в выбранной ячейке отобразится результат вычислений.





Рис. 9. Уменьшение диалогового окна *Мастера функций*

Редактировать формулу с функциями можно также в диалоговом окне *Мастера функций*. Для этого следует установить курсор в формуле на имя нужной функции и щелкнуть кнопку . В результате откроется окно *Аргументы функции* (см. рис. 8), в котором можно выполнить необходимые изменения.

В Excel существуют и другие способы вставки функций в формулу. Например, первый шаг *Мастера функций* (выбор функции) можно выполнить,

выбрав требуемую функцию из соответствующей категории на вкладке **Формулы** в группе Библиотека функций (см. рис. 6). В результате сразу откроется окно второго шага *Мастера функций* для ввода аргументов функции.

Кроме того, для ускорения ввода наиболее часто применяемых пользователем встроенных функций в Excel постоянно обновляется список из 10 последних набранных функций в обратном хронологическом порядке. После ввода в ячейку символа «=», определяющего начало формулы, в поле адреса ячейки появляется имя последней использованной функции (ее можно выбрать щелчком в этом поле), а в выпадающем списке – 10 последних (см. рис. 6), где можно выбрать требуемую функцию. Если среди них нужная функция отсутствует, ее можно найти, выбрав опцию *Другие функции*.

2.5. Кнопка Автосумма

В Excel имеется специальный инструмент для быстрого ввода наиболее часто используемых функций – кнопка **Σ Автосумма** на вкладках ленты **Главная** → [Редактирование] (рис. 10) и **Формулы** → [Библиотека функций]. Применение этого режима имеет следующие особенности.

1. В отличие от обычного ввода формулы при использовании кнопки Автосумма знак «=» в начале формулы набирать не нужно (он вводится автоматически).

2. Если результат требуется получить в ячейке, *продолжающей ряд исходных данных*, то нужно:

Способ 1:

- выделить все ячейки с исходными данными;
- в раскрывающемся списке кнопки Автосумма

выбрать нужную функцию;

Способ 2:

- выделить результирующую ячейку;
- в списке кнопки Автосумма выбрать нужную функцию, проверить автоматически выделенный диапазон ячеек для суммирования (при необходимости изменить его);
- нажать клавишу *Enter*.

Способ 2 применяется также, если ячейка результата *не продолжает ряд исходных данных* (находится в произвольном месте листа).




Рис. 10. Меню кнопки Автосумма

Если необходимо выполнить суммирование данных, то на этапе выбора нужной функции достаточно просто щелкнуть кнопку **Автосумма** или нажать комбинацию клавиш **Alt + =**.

Пример 2. В таблице, представленной на рис. 11, вычислить итоги по строкам и столбцам (заполнить ячейки в строке и столбце «Всего»).

	A	B	C	D	E	F	G
1	Ведомость на получение стипендии						
2	№	Фамилия И.О.	Сентябрь	Октябрь	Ноябрь	Декабрь	Всего
3	1	Авдеев К.В.	1600	1800	2000	2200	
4	2	Белова А.С.	1600	1800	2000	2200	
5	3	Иванова И.П.	1600	1800	2000	2200	
6	4	Котов Н.В.	1600	1800	2000	2200	
7	5	Смирнов С.Н.	1600	1800	2000	2200	
8	Всего						

Рис. 11. Пример таблицы для применения автосуммирования

Порядок действий:

1) выделить диапазон ячеек **C3:G8**, который содержит все данные для суммирования и ячейки для вывода результатов – пустую строку «Всего» и пустой столбец «Всего»;

2) щелкнуть кнопку **Автосумма**. В результате одновременно будут вычислены все итоги по строкам и столбцам.

Аналогичные действия выполняются, если нужно просуммировать данные только по строкам или по столбцам.

2.6. Отображение результатов вычислений в строке состояния

Excel позволяет выполнять простейшие вычисления, не вводя формулы в ячейки. При выделении на листе Excel ячеек с числовыми значениями в правой части строки состояния (нижней строки окна Excel) отображаются среднее арифметическое, количество и сумма этих значений (рис. 12). Перечень функций, значения которых для выделенных ячеек отображаются в строке состояния, можно дополнить или изменить. Для этого нужно щелкнуть правой кнопкой мыши в любом месте строки состояния и в открывшемся списке *Настройка строки состояния* выбрать нужные функции.

	A	B	C	D	E	F
1	Исходные данные		Результат			
2	3	7	31			
3	5	11	36			
4	-	-	0,86			
5						

Среднее: 13,40873016	Количество: 9	Сумма: 93,86111111
----------------------	---------------	--------------------

Рис. 12. Параметры строки состояния Excel

2.7. Абсолютные и относительные ссылки

В зависимости от выполняемых задач в формулах Excel применяют разные типы ссылок на ячейки – относительные, абсолютные или смешанные. Различие между ними проявляется при копировании или перемещении формул.

При вводе в формуле ссылки на ячейку Excel по умолчанию определяет ее как *относительную*, т. е. воспринимает не конкретный адрес этой ячейки, а ее смещение на заданное число столбцов и строк относительно той ячейки, где находится формула. Относительные ссылки обозначаются обычным образом, например, A1, K64. При копировании или перемещении формулы входящие в нее относительные ссылки автоматически изменяются в соответствии с относительным расположением исходной и результирующей ячеек с формулой. Например, в формуле в ячейке B2 имеется ссылка на ячейку A3, т. е. в относительном представлении это означает, что ссылка указывает на ячейку, которая располагается на один столбец левее и на одну строку ниже ячейки с формулой. Если формула будет скопирована в другую ячейку, то такое указание ссылки (взаимное расположение ячеек) сохранится. В нашем примере при копировании формулы из ячейки B2 в ячейку D4 имеющаяся в формуле ссылка на ячейку A3 изменится на ссылку на ячейку C5.

Абсолютная ссылка на ячейку в формуле всегда указывает на конкретную ячейку, расположенную строго в определенном месте рабочего листа. В абсолютных ссылках обозначения столбцов и строк предваряются символом \$, например, \$A\$1, \$K\$64. При копировании или перемещении формулы входящие в нее абсолютные ссылки не изменяются.

При комбинировании в одной ссылке относительной и абсолютной адресации получаются *смешанные* ссылки, позволяющие при копировании или перемещении формул изменять ссылки на столбцы, но сохранять ссылку на строку постоянной, либо изменять ссылки на строки, сохраняя ссылку на столбец постоянной. В смешанных ссылках только одна часть адреса (номер столбца или номер строки) указывается абсолютной, например, D\$16 (относительная ссылка по столбцу и абсолютная по строке) или \$F7 (абсолютная ссылка по столбцу и относительная по строке).

Для *изменения способа адресации ячеек* при вводе или редактировании формулы нужно установить курсор на ссылку на ячейку и нажать клавишу F4. При каждом нажатии F4 происходит последовательное изменение адресов: A1, \$A\$1, A\$1, \$A1 (для ячейки A1).

Копирование или перемещение формул можно осуществлять с помощью буфера обмена так же, как и любой другой информации, применяя различные способы выполнения команд *Вырезать*, *Копировать*, *Вставить* аналогично обычным приемам работы в операционной системе Windows и ее приложениях. При этом в результирующих формулах нужно обязательно проверять правильность автоматического изменения ссылок на ячейки.

Если однотипная формула повторяется для смежных ячеек, то целесообразно применять *автозаполнение формулами*: ввести формулу в первую результирующую ячейку, затем скопировать ее в другие ячейки, протянув за маркер заполнения. Другой (менее удобный) способ автозаполнения: выделить все ячейки с однотипной формулой, ввести формулу, в конце ввода нажать *Ctrl+Enter*.

Чтобы скопировать текст формулы из активной ячейки в другую без изменения ссылок, т. е. получить точную копию формулы, нужно произвести следующие действия:

- в режиме редактирования выделить исходную формулу и любым способом скопировать ее в буфер обмена;

- нажать клавишу *Esc*, чтобы выйти из режима редактирования активной ячейки;

- перейти в новую ячейку и вставить в нее формулу из буфера обмена.

Этот же способ можно применить, чтобы скопировать только часть формулы для использования ее в другой формуле.

Рассмотрим применение разных типов ссылок на конкретных примерах.

Пример 3. В таблице, представленной на рис. 13, заполнить выделенные серой заливкой ячейки D2:E5 – вычислить затраты на приобретение товара каждого вида и соответствующие доли затрат от общей стоимости.

	A	B	C	D	E
1	Товар	Количество, шт.	Цена за шт., руб.	Затраты, руб.	Доля затрат, %
2	Монитор	15	4500	=B2*C2	=D2/\$D\$5*100
3	Принтер	5	3700	=B3*C3	=D3/\$D\$5*100
4	Видеокарта	3	1900	=B4*C4	=D4/\$D\$5*100
5	Всего			=СУММ(D2:D4)	=СУММ(E2:E4)

а

	A	B	C	D	E
1	Товар	Количество, шт.	Цена за шт., руб.	Затраты, руб.	Доля затрат, %
2	Монитор	15	4500	67500	73,6
3	Принтер	5	3700	18500	20,2
4	Видеокарта	3	1900	5700	6,2
5	Всего			91700	100,0

б

Рис. 13. Вид рабочего листа Excel для примера 3 в режиме показа формул (а) и в обычном режиме просмотра (б)

Порядок действий:

1) ввести исходные данные в столбцы А, В, С (рис. 13, а). Для наглядности работы включить режим показа формул;

2) ввести формулы в столбец D (см. рис. 13, а) следующим образом:

в ячейку D2 ввести формулу =B2*C2;

с помощью маркера заполнения скопировать формулу из ячейки D2 в ячейки D3:D4. Так как исходная формула содержит относительные ссылки, то при копировании они автоматически изменяются в соответствии с новым положением формулы;

в ячейке D5 с помощью кнопки Автосумма вычислить сумму значений ячеек D2:D4;

3) ввести формулы в столбец E (см. рис. 13, а) следующим образом:

в ячейку E2 ввести формулу =D2/\$D\$5*100;

с помощью маркера заполнения скопировать формулу из ячейки E2 в ячейки E3:E4. Так как в исходной формуле ссылка на ячейку D5 задана абсолютной, то при копировании ее адрес не изменится, а относительная ссылка на ячейку D2 будет автоматически пересчитана;

4) скопировать формулу из ячейки D5 в ячейку E5 с помощью маркера заполнения;

5) для отображения результатов вычислений отключить режим показа формул (рис. 13, б).

Пример 4. Определить формулу и значение в ячейке D5 (рис. 14, а) после копирования в нее ячейки B2.

Решение задач такого типа осуществляют в два этапа. На первом этапе, не обращая внимания на саму формулу, определяют схему ее перемещения от исходной ячейки к результирующей. На втором этапе, наоборот, абстрагируясь от места расположения исходной и результирующей формул, анализируют непосредственно текст формулы и определяют в ней изменение относительных адресов ячеек в соответствии со схемой перемещения (абсолютные адреса ячеек в формуле остаются неизменными).

Порядок действий для решения примера 4:

1) составить схему перемещения ячейки B2 в ячейку D5 – на два столбца вправо и на три строки вниз (рис. 14, б);

2) исходная формула в ячейке B2 имеет вид: $=A2*\$C\1 (рис. 14, в). В соответствии со схемой перемещения относительная ссылка на ячейку A2 автоматически изменится на ссылку C5 (на два столбца вправо и на три строки вниз), абсолютная ссылка $\$C\1 не изменится, следовательно, в ячейке D5 будет формула $=C5*\$C\1 , расчет по которой даст в ответе число 8 (рис. 14, г).

Пример 5. Определить формулу и значение в ячейке E5 (рис. 15, а) после копирования в нее ячейки B3.

Порядок действий:

1) составить схему перемещения ячейки B3 в ячейку E5 (рис. 15, б) – на три столбца вправо и на две строки вниз;

2) исходная формула в ячейке B3 имеет вид: $=B4+B\$2-\$C3$ (рис. 15, в). В соответствии со схемой перемещения относительная ссылка на ячейку B4 автоматически изменится на ссылку E6, смешанная ссылка B\$2 – на E\$2,

смешанная ссылка \$C3 – на \$C5. Следовательно, в ячейке E5 будет формула =E6+E\$2-\$C5, расчет по которой даст в ответе число -7 (рис. 15, г).




Рис. 14. Копирование ячейки B2 в ячейку D5: исходные данные (а); схема перемещения (б); режим показа формул (в); результат (г)




Рис. 15. Копирование ячейки B3 в ячейку E5: исходные данные (а); схема перемещения (б); режим показа формул (в); результат (г)

Пример 6. Определить формулу и значение в ячейке Е3 (рис. 16, а) после копирования в нее ячейки С6.


Порядок действий:

1) составить схему перемещения ячейки С6 в ячейку Е3 (рис. 16, б) – на два столбца вправо и на три строки вверх;

2) исходная формула в ячейке С6 имеет вид: $=\$B\$3*C4-\$B7$ (рис. 16, в). В соответствии со схемой перемещения (см. рис. 16, б) абсолютная ссылка $\$B\3 не изменится, относительная ссылка на ячейку С4 автоматически изменится на ссылку Е1, смешанная ссылка $\$B7$ – на $\$B4$. Следовательно, в ячейке Е3 будет формула $=\$B\$3*E1-\$B4$, расчет по которой даст в ответе число 4 (рис. 16, г).

C6		fx = $=\$B\$3*C4-\$B7$			
	A	B	C	D	E
1		8	2	9	3
2		4	7	1	5
3		3	2	13	?
4		5	1	4	14
5		-11	15	1	-5
6		2	-5	8	3
7		8	10	21	16

а



б

CVMM		fx = $=\$B\$3*C4-\$B7$			
	A	B	C	D	E
1		8	2	9	3
2		4	7	1	5
3		3	2	13	?
4		5	1	4	14
5		-11	15	1	-5
6		2	$=\$B\$3*C4-\$B7$	8	3
7		8	10	21	16

в

E3		fx = $=\$B\$3*E1-\$B4$			
	A	B	C	D	E
1		8	2	9	3
2		4	7	1	5
3		3	2	13	4
4		5	1	4	14
5		-11	15	1	-5
6		2	-5	8	3
7		8	10	21	16

г

Рис. 16. Копирование ячейки С3 в ячейку Е3: исходные данные (а); схема перемещения (б); режим показа формул (в); результат (г)

2.8. Работа с массивами данных

Массив данных представляет собой набор значений, сгруппированных по строкам и столбцам. Примерами одномерных массивов в математике являются векторы, двумерных – матрицы. Вычисления с использованием массивов в Excel имеют ряд особенностей и осуществляются в следующем порядке:

1) ввести значения элементов исходного массива в ячейки листа Excel;
2) выделить диапазон ячеек (или одну ячейку в зависимости от решаемой задачи), где требуется получить результат вычислений;

3) ввести формулу и нажать комбинацию клавиш *Ctrl + Shift + Enter*. При этом формула автоматически будет заключена в фигурные скобки и скопирована во все выделенные ячейки. Эти действия выполняются всегда, если нужно одновременно воздействовать на набор значений. Когда результат расчета предполагается только в одной ячейке, то вместо указанной комбинации клавиш в диалоговом окне можно нажать клавишу *Enter* или *OK*.

Для исключения ошибочных ситуаций при работе с массивами перед нажатием комбинации клавиш *Ctrl + Shift + Enter* курсор всегда следует устанавливать в строку формул.

В некоторых случаях при вычислениях в ячейке результата Excel выдает ошибку #ЗНАЧ. При работе с массивами наиболее частой причиной такой ошибки является то, что для работы с элементами массива была введена простая формула, т. е. после ввода формулы вместо комбинации клавиш *Ctrl + Shift + Enter* нажата клавиша *Enter* или кнопка *OK*. Исправить это можно следующим образом: выделить диапазон результирующих ячеек, перейти в строку формул и нажать комбинацию клавиш *Ctrl + Shift + Enter*.

Отдельные элементы результирующего массива изменять нельзя.

Для работы с массивами в Excel предусмотрен ряд встроенных функций, в том числе:

- МОПРЕД – возвращает определитель матрицы;
- МОБР – возвращает обратную матрицу;
- ТРАНСП – транспонирует матрицу (меняет строки и столбцы местами).

При транспонировании матрицы произвольного размера $n \times m$ следует иметь в виду, что результирующая матрица будет иметь размер $m \times n$, и с учетом этого выделять диапазон ячеек при создании формулы.

Пример 7. Для исходной матрицы размером 3×3 , расположенной в ячейках A2:C4 (рис. 17), вычислить определитель, обратную и транспонированную матрицы. Получить матрицу, каждый элемент которой равен квадрату соответствующего элемента исходной матрицы.

	A7			f_x	{=МОБР(A2:C4)}						
	A	B	C	D	E	F	G	H	I	J	K
1	Исходная матрица										
2	3	6	0								
3	4	2	-4		Определитель =		-174				
4	3	2	7								
5											
6	Обратная матрица				Транспонированная матрица				Элементы в квадрате		
7	-0,12644	0,241379	0,137931		3	4	3		9	36	0
8	0,229885	-0,12069	-0,06897		6	2	2		16	4	16
9	-0,01149	-0,06897	0,103448		0	-4	7		9	4	49

Рис. 17. Пример работы с матрицей

Порядок действий:

вычисление определителя матрицы:

- 1) выделить ячейку результата, например, G3;
- 2) запустить *Мастер функций*, выбрать функцию МОПРЕД;
- 3) в диалоговом окне функции МОПРЕД в поле *Массив* указать ячейки с элементами исходной матрицы (A2:C4);
- 4) нажать клавишу *ОК*. В итоге формула примет окончательный вид =МОПРЕД(A2:C4), а в результирующей ячейке G3 появится вычисленное значение -174 (см. рис. 17);

вычисление обратной матрицы:

- 1) выделить диапазон ячеек, куда нужно поместить результирующую матрицу, например, A7:C9. При этом количество выделенных ячеек и их положение должны строго соответствовать объему результирующих данных;
- 2) запустить *Мастер функций*, выбрать функцию МОБР;
- 3) в диалоговом окне функции МОБР в поле *Массив* указать ячейки с элементами исходной матрицы (A2:C4). При этом в строке формул будет создана формула =МОБР(A2:C4);
- 4) установить курсор в строку формул;
- 5) нажать комбинацию клавиш *Ctrl + Shift + Enter*. В итоге формула будет заключена в фигурные скобки и примет окончательный вид {=МОБР(A2:C4)}, а в результирующих ячейках A7:C9 появятся вычисленные значения (см. рис. 17);

вычисление транспонированной матрицы выполняется с использованием функции ТРАНСП аналогично вычислению обратной матрицы (см. рис. 17);

возведение элементов матрицы в квадрат:

1) выделить диапазон ячеек, куда нужно поместить результирующую матрицу, например, I7:K9.

2) ввести знак «=», выделить исходную матрицу (в строке формул появится текст =A2:C4);

3) установить курсор в строку формул и набрать ^2 (возведение в квадрат). При этом в строке формул будет создана формула =A2:C4^2;

4) нажать комбинацию клавиш *Ctrl + Shift + Enter*. В итоге формула будет заключена в фигурные скобки и примет окончательный вид {=A2:C4^2}, а в результирующих ячейках I7:K9 появятся вычисленные значения (см. рис. 17).

2.9. Функция СУММПРОИЗВ

Для повышения эффективности вычислительного процесса в Excel существуют функции, объединяющие в себе несколько действий одновременно. Например, функция СУММПРОИЗВ перемножает соответствующие элементы заданных диапазонов ячеек (массивов данных) и возвращает сумму полученных произведений. В библиотеке встроенных функций Excel она находится в категории *Статистические*.

Общая форма записи функции СУММПРОИЗВ:

СУММПРОИЗВ(Массив1; [Массив2]; [Массив3]; ...),

где Массив1, [Массив2], [Массив3], ... – диапазоны или массивы (от 2 до 255), соответствующие компоненты которых нужно сначала перемножить, а затем сложить полученные результаты.

Следует иметь в виду, что нечисловые элементы массивов функция СУММПРОИЗВ трактует как нулевые.

Пример 8. В таблице на рис. 18 представлены результаты тестирования студентов. В ячейках C3:G3 задано максимальное количество баллов за каждый тест, в ячейках C5:G9 – результаты выполнения студентами тестов в виде коэффициента (от 0 до 1 включительно). В графе «Общий балл» необходимо вычислить общее количество баллов для каждого студента, в графе «Округление» – округлить полученный результат до целых значений.

Порядок действий:

1) установить курсор в первую результирующую ячейку Н5;

2) запустить *Мастер функций*, выбрать функцию СУММПРОИЗВ и в открывшемся диалоговом окне этой функции ввести ее аргументы (рис. 19):

в поле *Массив1* выделить на рабочем листе первый диапазон ячеек С3:G3, значения которых будут участвовать в расчете. Строка максимального количества баллов является постоянной для всех студентов, поэтому здесь необходимо применить абсолютную адресацию (нажать клавишу F4), чтобы при копировании формулы для других студентов ссылки на эти ячейки не изменялись (таким образом, этот аргумент функции будет иметь вид \$C\$3:\$G\$3);

в поле *Массив2* выделить второй диапазон ячеек С5:G5, значения которых будут участвовать в расчете для студента Авдеева К.В.;

3) нажать кнопку *ОК*, вследствие чего в ячейке Н5 автоматически сформируется формула =СУММПРОИЗВ(\$C\$3:\$G\$3;C5:G5), и Excel отобразит в ней результат вычислений – 76,25 (общее количество баллов студента Авдеева К.В.);

4) установить курсор в результирующую ячейку I5;

5) запустить *Мастер функций*, выбрать функцию ОКРУГЛ и в открывшемся диалоговом окне этой функции ввести ее аргументы (рис. 20):

в поле *Число* – ссылку на ячейку Н5;

в поле *Число_разрядов* – 0;

6) нажать кнопку *ОК*, вследствие чего в ячейке I5 автоматически сформируется формула =ОКРУГЛ(Н5;0), и Excel отобразит в ней результат вычислений – 76;

7) выделить ячейки с формулами Н5:I5 и скопировать их в ячейки Н6:I9, протянув вниз за маркер заполнения.

Н5		fx =СУММПРОИЗВ(\$C\$3:\$G\$3;C5:G5)							
A	B	C	D	E	F	G	H	I	
1	№	Фамилия И.О.	Максимальное кол-во баллов за тест					Общий балл	Округ- ление
2			Тест1	Тест2	Тест3	Тест4	Тест5		
3			15	20	25	20	20		
4			Коэффициент выполнения теста						
5	1	Авдеев К.В.	0,98	0,75	0,67	0,80	0,69	76,25	76
6	2	Белова А.С.	0,23	0,33		0,27		15,45	15
7	3	Иванова И.П.	1,00	0,80	1,00	0,95	1,00	95,00	95
8	4	Котов Н.В.	0,60	0,52	0,64	0,83	0,77	67,40	67
9	5	Смирнов С.Н.	0,49	0,31	0,92	0,88		54,15	54

Рис. 18. Таблица результатов тестирования




Рис. 19. Диалоговое окно функции СУММПРОИЗВ




Рис. 20. Диалоговое окно функции ОКРУГЛ

2.10. Контрольные вопросы

- 1) Правила ввода формулы в ячейку.
- 2) Назовите типы адресов ячеек в Excel и их различия.
- 3) Как изменить способ адресации ячейки при вводе формулы?
- 4) С помощью каких встроенных функций можно вывести текущую дату?
- 5) Что такое Мастер функций?
- 6) Как ввести формулу с использованием кнопки Автосумма?
- 7) В чем заключаются особенности работы с массивом данных?

2.11. Задания

Задание 1. Ввод формул. Абсолютные и относительные ссылки.

1. Откройте Excel-шаблон.

2. Сохраните Excel-шаблон с именем *Фамилия_Excel-2* (здесь и далее вместо слова «Фамилия» укажите вашу фамилию) в личной папке. Все дальнейшие действия выполняйте в этом файле.

3. В свойствах файла в поле *Примечания* укажите свои фамилию и группу.

4. Создайте новый рабочий лист, назвав его «Лаб2_31».

5. В верхнем колонтитуле справа введите свой учебный шифр, в нижнем колонтитуле справа – фамилию и инициалы преподавателя.

6. В ячейке A1 введите текст «*Текущая дата*», в ячейке B1 с помощью функции СЕГОДНЯ выведите текущую дату.

7. В ячейке A2 введите текст «*Дата рождения*», в ячейке B2 – дату своего рождения в формате ДДММГГГГ.

8. В ячейке A3 введите текст «*Количество дней*», в ячейке B3 вычислите количество прожитых вами дней на текущую дату.

9. В ячейке D2 введите текст «*Юбилей*», в ячейке E2 введите формулу =ОКРУГЛВВЕРХ(B3; -3) для вычисления количества дней, соответствующего ближайшему тысячному юбилею со дня вашего рождения.

10. В ячейке D3 введите текст «*Дата*», в ячейке E3 вычислите дату, когда вы отметите свой ближайший тысячный юбилей.

11. Создайте таблицу в соответствии с образцом, приведенным на рис. 21: введите исходные данные в ячейки C7:D11;

к ячейкам E7:F12 примените заливку серым цветом и вычислите их значения с помощью формул с обязательным использованием относительных и абсолютных ссылок аналогично примеру 3 в подразд. 2.7.

12. Ячейки A13:B13 объедините в одну ячейку, в которую введите текст «Расчет с помощью функции СУММПРОИЗВ».

13. Ячейки C13:D13 объедините в одну ячейку, в которой на основе данных ячеек C7:C11 и D7:D11 вычислите общий размер платежа за коммунальные услуги с помощью функции СУММПРОИЗВ. Сравните полученные разными способами результаты в ячейках C13 и E12 и убедитесь в их идентичности.

14. На свободное место листа вставьте рисунок таблицы в режиме проверки формул и уменьшите его размер.

15. Сохраните содержимое рабочей книги.

C13 fx =СУММПРОИЗВ(C7:C11;D7:D11)						
	A	B	C	D	E	F
1	Текущая дата	28.03.2018				
2	Дата рождения	14.09.2000		Юбилей	7000	
3	Количество дней	6404		Дата	14.11.2019	
4						
5	Платежи за коммунальные услуги					
6	Ресурс	Ед. изм.	Расход, ед. изм.	Тариф, руб.	Размер платежа, руб.	Доля платежа, %
7	Холодная вода	м ³	5	15,29	76,45	1,8
8	Горячая вода	м ³	7	93,40	653,80	15,6
9	Отопление	Гкал	1,4	1648,16	2307,42	55,0
10	Газ	м ³	3	90,01	270,03	6,4
11	Электроэнергия	кВт·ч	250	3,56	890,00	21,2
12	Всего				4197,70	100,0
13	Расчет с помощью функции СУММПРОИЗВ		4197,70			

Рис. 21. Образец выполнения задания 1

Задание 2. Смешанные ссылки.

1. Откройте свою рабочую книгу *Фамилия_Excel-2*, создайте в ней новый рабочий лист с именем «Лаб2-32», на котором создайте таблицу умножения в соответствии с образцом, приведенным на рис. 22 (включая оформление – форматирование шрифта, толстые границы и др.), в следующем порядке:

с помощью автозаполнения введите сомножители от 1 до 9 по строкам и столбцам;

используя смешанные ссылки на ячейки с сомножителями, введите формулу в первую расчетную ячейку (для расчета 2×2) и убедитесь в правильности вычисления произведения ($2 \times 2 = 4$);

проверьте правильность введенной формулы: скопируйте ее в любую ячейку, например, для расчета 6×7 , и проверьте результат ($6 \times 7 = 42$). Если он неверен, измените формулу в первой расчетной ячейке, снова ее скопируйте и проверьте результат;

Таблица умножения								
1	2	3	4	5	6	7	8	9
2	4	6	8	10	12	14	16	18
3	6	9	12	15	18	21	24	27
4	8	12	16	20	24	28	32	36
5	10	15	20	25	30	35	40	45
6	12	18	24	30	36	42	48	54
7	14	21	28	35	42	49	56	63
8	16	24	32	40	48	56	64	72
9	18	27	36	45	54	63	72	81

Рис. 22. Образец выполнения задания 2

с помощью автозаполнения скопируйте формулу во все расчетные ячейки таблицы.

2. Сохраните содержимое рабочей книги.

Задание 3. Применение встроенных функций КОРЕНЬ и LOG.

1. Откройте свою рабочую книгу *Фамилия_Excel-2*, создайте в ней новый рабочий лист с именем «Лаб2-33», на котором выполните приведенные ниже задания.

	A	B	C	D
5	Комп.	3		
6	ИВ	17		
7			x	y(x)
8			3	0,021
9			20	0,057
10			37	0,069
11			54	0,077
12			71	0,082
13			88	0,086
14			105	0,090
15			122	0,093
16			139	0,096
17			156	0,098
18		S=	795	
19				
20		A=	79,5	

Рис. 23. Образец выполнения задания 3

2. В ячейку B5 введите число N , равное номеру своего компьютера, в ячейку B6 – заданный преподавателем номер вашего индивидуального варианта (далее – ИВ).

3. С помощью *формул* заполните ячейки C8:C17 с начальным значением в ячейке B5 и шагом, равным номеру ИВ, по образцу, показанному на рис. 23.

4. В ячейке C18 вычислите сумму чисел S в ячейках C8:C17, в ячейке C20 – среднее арифметическое тех же чисел A .

5. Ячейки D8:D17 заполните значениями функции $y(x)$, вычислив их по формуле

$$y(x) = \frac{2\sqrt{x} + S}{A^2} \log_S x, \text{ где } x - \text{соответствующие значения из ячеек C9:C18. Для вычисления квадратного корня используйте встроенную функцию КОРЕНЬ}(x), \text{ для логарифма - LOG}(x; S).$$

6. Установите в ячейке C20 числовой формат с одним десятичным знаком, а в ячейках D8:D17 – с тремя десятичными знаками.

7. В ячейке A6 создайте примечание, в котором введите свою фамилию.

8. Сохраните содержимое рабочей книги.

Задание 4. Работа с массивом данных.

1. Откройте свою рабочую книгу *Фамилия_Excel-2*, создайте в ней новый рабочий лист с именем «Лаб2-34», на котором выполните приведенные ниже задания.

2. Введите произвольную исходную матрицу размером 4×4 (первый элемент матрицы должен быть равен номеру ИВ).

3. Для исходной матрицы вычислите:

- определитель;
- обратную матрицу;
- транспонированную матрицу;
- матрицу, каждый элемент которой равен квадрату соответствующего элемента исходной матрицы.

Оформите работу по образцу, представленному на рис. 17.

4. В ячейке с первым элементом исходной матрицы создайте примечание, в котором введите свою фамилию и номер ИВ.

5. Сохраните содержимое рабочей книги.

Задание 5. Преобразования чисел в разных системах счисления.

1. Откройте свою рабочую книгу *Фамилия_Excel-2*, создайте в ней новый рабочий лист с именем «Лаб2-35», на котором в соответствии с номером ИВ выполните три задания (табл. 3) по преобразованию заданных чисел из исходной системы счисления с основанием $CC_{исх}$ в результирующую систему счисления с основанием $CC_{рез}$ с использованием встроенных функций Excel из категории *Инженерные*. Оформите работу по образцу, представленному на рис. 24.

Таблица 3

Варианты заданий по теме «Системы счисления»

Вариант	Задание CC1			Задание CC2			Задание CC3		
	Число	$CC1_{исх}$	$CC1_{рез}$	Число	$CC2_{исх}$	$CC2_{рез}$	Число	$CC3_{исх}$	$CC3_{рез}$
1	110001	2	8	7B	16	10	37	10	2
2	1216	10	16	101010	2	8	425	8	10
3	706	8	10	11E	16	8	110001	2	16
4	6D	16	2	47	10	2	307	8	16
2	491	10	8	101011	2	16	9A	16	2
5	226	8	16	361	10	8	110010	2	10
6	111110	2	8	1A2	16	10	517	8	16
7	2A	16	10	110011	2	8	32	10	2
8	48	10	2	412	8	10	101111	2	16
9	234	8	16	100101	2	16	F8	16	10
10	217	10	8	354	8	10	110100	2	8
11	100010	2	16	32	10	2	572	8	16
12	E2	16	10	611	8	16	19	10	2
13	564	8	2	582	10	8	100111	2	16
14	226	10	16	100010	2	16	235	8	10
15	110101	2	8	127	8	10	FB	16	2

	A	B	C	D	E	F
1	Вариант	Задание	Число	СС _{исх}	СС _{рез}	Результат
2	0	СС1	183	10	16	В7
3		СС2	1011	2	8	13
4		СС3	247	8	10	167

Рис. 24. Образец выполнения задания 5

Задание 6. Решение задач по теме «Измерение информации».

1. Откройте свою рабочую книгу Фамилия_Excel-2, создайте в ней новый рабочий лист с именем «Лаб2-зб», на котором выполните приведенные ниже задания.

2. В соответствии с номером ИВ решите задачу 5 по теме «Измерение информации» [5, с. 24 – 31]. Для этого используйте шаблон таблицы, представленный на рис. 25.

	A	B	C	D	E
1	Шаблон для решения задач по теме "Измерение информации"				
2	Вариант		Задача №		
3	Условие задачи:				
4					
5	Событие	Количество событий	Вероятность события	$1 / p_i$	Количество информации в сообщении о событии
6		k_i	$p_i = k_i / N$		$I_i = \log_2(1 / p_i)$, бит
7					
8					
9					
10					
11	Всего			–	–
12		$N = \sum k_i$	$\sum p_i = 1$		

Рис. 25. Шаблон таблицы для решения задач по теме «Измерение информации»

В ячейки шаблона введите:

B2 – номер ИВ;

D2 – номер задачи (5);

A4 – формулировку задачи.

В ячейках A7:E10; B11; C11 выполните следующие действия:

в соответствии с формулировкой задачи введите исходные данные и выделите их шрифтом красного цвета;

ячейку результата выделите заливкой желтого цвета;

введите формулы только в те ячейки, которые необходимы для расчета.

3. В ячейку A5 добавьте примечание, в котором введите свою фамилию и номер ИВ.

4. На этом же листе ниже (начиная с ячейки A15) в соответствии с номером ИВ аналогично решите задачу 6 по теме «Измерение информации» [5, с. 24 – 31].

5. Сохраните содержимое рабочей книги.

Библиографический список

1. Гаврилов М. В. Информатика и информационные технологии: Учебник / М. В. Гаврилов, В. А. Климов. М.: Юрайт, 2016. 383 с.

2. Патеюк А. Г. Обработка данных в табличном процессоре Microsoft Excel 2010. Часть 1. Основы вычислений в электронных таблицах: Учебное пособие / А. Г. Патеюк/ Омский гос. ун-т путей сообщения. Омск, 2014. 58 с.

3. Насташук Н. А. Обработка данных в табличном процессоре Microsoft Excel 2010. Часть 2. Работа с диаграммами и списками в электронных таблицах: Учебное пособие / Н. А. Насташук/ Омский гос. ун-т путей сообщения. Омск, 2014. 87 с.

4. Сидорова Е. А. Лабораторный практикум по работе в табличном процессоре Microsoft Excel 2010. Часть 1. Ввод, редактирование и форматирование данных: Практикум / Е. А. Сидорова, С. П. Железняк/ Омский гос. ун-т путей сообщения. Омск, 2018. 33 с.

5. Сидорова Е. А. Измерение информации: Учебно-методическое пособие / Е. А. Сидорова, Т. А. Манохина, С. П. Железняк/ Омский гос. ун-т путей сообщения. Омск, 2017. 32 с.

Учебное издание

СИДОРОВА Елена Анатольевна,
ЖЕЛЕЗНЯК Светлана Петровна

ЛАБОРАТОРНЫЙ ПРАКТИКУМ ПО РАБОТЕ
В ТАБЛИЧНОМ ПРОЦЕССОРЕ MICROSOFT EXCEL 2010

Часть 2

СОЗДАНИЕ ФОРМУЛ. ВСТРОЕННЫЕ ФУНКЦИИ

Практикум

Редактор Н. А. Майорова

Подписано в печать 03.04.2018. Формат 60 × 84 ¹/₁₆.
Офсетная печать. Бумага офсетная. Усл. печ. л. 2,3. Уч.-изд. л. 2,5.
Тираж 500 экз. Заказ .

**

Редакционно-издательский отдел ОмГУПСа
Типография ОмГУПСа

*

644046, г. Омск, пр. Маркса, 35