

Е. А. СИДОРОВА, С. П. ЖЕЛЕЗНЯК

**ЛАБОРАТОРНЫЙ ПРАКТИКУМ ПО РАБОТЕ
В ТАБЛИЧНОМ ПРОЦЕССОРЕ MICROSOFT EXCEL 2010**

ЧАСТЬ 3

ВЫЧИСЛЕНИЯ С ПРОВЕРКОЙ УСЛОВИЙ

ОМСК 2018

Министерство транспорта Российской Федерации
Федеральное агентство железнодорожного транспорта
Омский государственный университет путей сообщения

Е. А. Сидорова, С. П. Железняк

ЛАБОРАТОРНЫЙ ПРАКТИКУМ ПО РАБОТЕ
В ТАБЛИЧНОМ ПРОЦЕССОРЕ MICROSOFT EXCEL 2010

Часть 3

ВЫЧИСЛЕНИЯ С ПРОВЕРКОЙ УСЛОВИЙ

Утверждено методическим советом университета
в качестве практикума к выполнению самостоятельной и
лабораторных работ

Омск 2018

УДК 004.67(075.8)
ББК 32.973я73
С34

Лабораторный практикум по работе в табличном процессоре Microsoft Excel 2010. Часть 3. Вычисления с проверкой условий /
Е. А. Сидорова, С. П. Железняк; Омский гос. ун-т путей сообщения. Омск, 2018. 30 с.

Лабораторный практикум состоит из трех частей. В третьей части рассмотрены особенности создания формул в табличном процессоре Microsoft Excel 2010 с применением встроенных функций, содержащих проверку выполнения условий. Представлены примеры реализации наиболее часто встречающихся операций и практические задания.

Практикум предназначен для студентов и аспирантов очной и заочной форм обучения всех направлений подготовки (специальностей), изучающих дисциплины информационного профиля: «Информатика», «Информационные технологии», «Компьютерные технологии и информатика», «Информационные системы и базы данных» и др., а также может быть использован для самостоятельной работы любых категорий пользователей персонального компьютера.

Библиогр.: 4 назв. Табл. 2. Рис. 16.

Рецензенты: доктор техн. наук, профессор В. Н. Горюнов;
доктор техн. наук, профессор А. А. Кузнецов.

ОГЛАВЛЕНИЕ

Введение	5
Лабораторная работа 3. Вычисления с проверкой условий.....	6
3.1. Встроенные функции Excel с проверкой условий	6
3.2. Функция ЕСЛИ	7
3.3. Вложенные функции	10
3.4. Функция СЧЁТЕСЛИ.....	11
3.5. Функция СЧЁТЕСЛИМН	12
3.6. Функция СУММЕСЛИ	13
3.7. Организация разветвлений с помощью вложенных функций ЕСЛИ	16
3.8. Логические функции НЕ, И, ИЛИ	18
3.9. Контрольные вопросы.....	21
3.10. Задания	21
Задание 1. Вычисления с проверкой условий	21
Задание 2. Суммирование с проверкой условия	23
Задание 3. Вложенные функции ЕСЛИ	25
Задание 4. Работа с логическими функциями	28
Библиографический список.....	29

ВВЕДЕНИЕ

Табличный процессор Microsoft Excel 2010 (далее – Excel) представляет собой удобный инструмент для автоматизации вычислений в различных сферах деятельности человека. Большой набор стандартных встроенных функций позволяет успешно решать задачи разного уровня сложности, обрабатывать и управлять большими массивами информации, обеспечивает получение достоверных результатов.

В лабораторной работе 3 рассматриваются вопросы эффективного применения функций обработки данных, удовлетворяющих некоторому условию, а также логических функций, представлены задания для самостоятельного выполнения студентом. После завершения лабораторной работы студент предъявляет преподавателю файл с выполненным заданием.

Библиографический список, представленный в конце практикума, содержит литературу для углубленного изучения материала по рассматриваемой тематике.

ВЫЧИСЛЕНИЯ С ПРОВЕРКОЙ УСЛОВИЙ

Цель работы:

- 1) изучение возможностей Excel для реализации вычислений, содержащих проверку выполнения условий;
- 2) освоение способов применения встроенных функций, содержащих проверку выполнения условий, и логических функций;
- 3) закрепление навыков обработки данных, удовлетворяющих заданным условиям.

3.1. Встроенные функции Excel с проверкой условий

В научной и производственной деятельности часто приходится решать задачи по отбору и обработке данных, удовлетворяющих заданным условиям. В Excel для этого имеется ряд встроенных функций, в том числе:

ЕСЛИ – осуществляет проверку выполнения заданного условия в ячейках;

СЧЁТЕСЛИ, СЧЁТЕСЛИМН – подсчитывают количество непустых ячеек, значения которых удовлетворяют одному или множеству проверяемых условий;

СУММЕСЛИ – вычисляет сумму значений ячеек в заданном диапазоне, удовлетворяющих проверяемому условию.

Рассмотрим особенности применения этих функций на примере таблицы результатов сессии (рис. 1), в которой нужно заполнить выделенные заливкой ячейки:

- 1) в ячейках I3:I9 вывести фамилии студентов, имеющих средний балл 4 и выше;
- 2) в ячейках J3:J9 назначить стипендию в размере 5000 руб. студентам, сдавшим все экзамены на 4 и 5;
- 3) в ячейках D10:G10 подсчитать общее количество пятерок по каждой учебной дисциплине;
- 4) в ячейках J11 и J12 вычислить суммы стипендии, назначенной студентам указанных групп.

Порядок действий для решения каждой из этих задач представлен ниже соответственно в примерах 1 – 4.

	A	B	C	D	E	F	G	H	I	J
1	№ п/п	Фамилия И.О. студента	Группа	Оценки по дисциплинам					Список студентов (ср. балл ≥ 4)	Стипендия, руб.
2				Математика	Физика	Информатика	История	Средний балл		
3	1	Авдеев К.В.	47а	4	5	3	3	3,75		
4	2	Белова А.С.	47б	4	4	5	5	4,50		
5	3	Ерофеев Д.И.	47а	5	5	4	5	4,75		
6	4	Иванов И.П.	47б	5	5	5	5	5,00		
7	5	Котова Н.В.	47б	5	4	3	5	4,25		
8	6	Морозов Ф.Д.	47а	3	3	3	3	3,00		
9	7	Федюнин И.Н.	47б	5	4	4	5	4,50		
10	Количество 5 по дисциплинам									
11								Стипендия по группам, руб.	47а	
12									47б	

Рис. 1. Исходная таблица результатов сессии

3.2. Функция ЕСЛИ

Функция ЕСЛИ проверяет, выполняется ли условие, и возвращает одно значение, если оно выполняется, и другое значение, если оно не выполняется. В библиотеке встроенных функций Excel функция ЕСЛИ находится в категории *Логические*.

Общая форма записи функции ЕСЛИ:

ЕСЛИ(Лог_выражение; Значение_если_истина; Значение_если_ложь),

где Лог_выражение – любое значение или выражение, которое при вычислении дает значение *Истина* или *Ложь* (здесь записывается проверяемое условие);

Значение_если_истина – значение, которое возвращается, если Лог_выражение имеет значение *Истина* (здесь записывается результат, который функция должна вывести, если проверяемое условие выполняется). Если этот аргумент не задан, то возвращается значение *Истина*.

Значение_если_ложь – значение, которое возвращается, если Лог_выражение имеет значение *Ложь* (здесь записывается результат, который функция должна вывести, если проверяемое условие не выполняется). Если этот аргумент не задан, то возвращается значение *Ложь*.

Пример 1. В ячейках I3:I9 (см. рис. 1) вывести фамилии студентов, имеющих средний балл 4 и выше.

Порядок действий:

1. Установить курсор в первую результирующую ячейку I3.

2. Запустить Мастер функций, выбрать функцию ЕСЛИ и в открывшемся диалоговом окне этой функции ввести ее аргументы (рис. 2):

в поле Лог_выражение – проверяемую ячейку и заданное условие ($H3 \geq 4$), при этом Excel справа от этого поля отобразит слово ЛОЖЬ, означающее, что проверяемое условие в данном случае не выполняется, поскольку значение в ячейке H3 равно 3,75;




Рис. 2. Диалоговое окно функции ЕСЛИ для примера 1

в поле Значение_если_истина – результат, который следует вывести в ячейке I3, если Лог_выражение окажется истинным. Здесь указывается ссылка на ячейку B3, содержимое которой (фамилию студента) Excel должен отобразить в результирующей ячейке I3, если проверяемое условие будет выполняться;

в поле Значение_если_ложь – результат, который следует вывести в ячейке I3, если Лог_выражение окажется ложным. В нашем примере для студентов, имеющих средний балл менее 4, фамилии отображать не нужно, т. е. соответствующие ячейки визуально должны быть пустыми. Однако если поле Значение_если_ложь в диалоговом окне оставить пустым, то при невыполнении условия, заданного в поле Лог_выражение, Excel отобразит в результирующей ячейке I3 слово ЛОЖЬ. Чтобы этого избежать, надо набрать в поле Значение_если_ложь пробел (или текст, которым нужно сопроводить вывод соответствующего результата).

3. Нажать кнопку ОК, вследствие чего в ячейке I3 автоматически сформируется формула =ЕСЛИ(H3>=4; B3; " "), и Excel отобразит в ней результат вычислений – ячейка визуально останется пустой, так как проверяемое условие

для студента Авдеева К. В. не выполняется (его средний балл ниже 4). Следует отметить, что в созданной формуле пробел заключен в кавычки, как это должно быть с любыми текстовыми данными, являющимися аргументами функций. При необходимости Excel ставит кавычки автоматически, в том числе и в полях ввода аргументов в диалоговых окнах функций.

4. Скопировать формулу из ячейки I3 в ячейки I4:I9 с помощью маркера заполнения.

Результат выполнения примера 1 в ячейках I3:I9 представлен в подразд. 3.6 (стр. 15).

Пример 2. В ячейках J3:J9 назначить стипендию в размере 5000 руб. студентам, сдавшим все экзамены на 4 и 5.

Порядок действий:

1. Установить курсор в первую результирующую ячейку J3.

2. Запустить Мастер функций, выбрать функцию ЕСЛИ и в открывшемся диалоговом окне этой функции ввести ее аргументы (рис. 3):

в поле **Лог_выражение** – проверяемое условие. Так как ни одна оценка студента в ячейках D3:G3 не должна быть менее 4, достаточно проверить минимальную из них, определив ее значение с помощью функции МИН. Тогда в поле **Лог_выражение** необходимо набрать условие $\text{МИН}(D3:G3) \geq 4$ (при этом для ввода диапазона ячеек D3:G3 достаточно выделить его мышью);

в поле **Значение_если_истина** – число 5000;

в поле **Значение_если_ложь** – число 0.




Рис. 3. Диалоговое окно функции ЕСЛИ для примера 2

3. Нажать кнопку ОК, вследствие чего в ячейке J3 автоматически сформируется формула `=ЕСЛИ(МИН(D3:G3)>=4; 5000; 0)`, и Excel отобразит в ней результат вычислений – число 0, так как у студента Авдеева К. В. есть тройки.

4. Скопировать формулу из ячейки J3 в ячейки J4:J9 с помощью маркера заполнения.

Результат выполнения примера 2 в ячейках J3:J9 представлен в подразд. 3.6 (стр. 15).

3.3. Вложенные функции


В рассмотренном выше примере 2 созданная формула объединяет в себе сразу две встроенные функции Excel (ЕСЛИ и МИН) и представляет собой пример *вложенных* функций. При создании таких формул следует соблюдать правило – на любом уровне вложенности¹ «внутренняя» (вложенная) функция со всеми своими аргументами должна полностью входить в состав одного аргумента «внешней» для нее функции.

Ввод формул с вложенными функциями ничем не отличается от ввода обычных формул, однако применение для этого Мастера функций имеет ряд особенностей. Рассмотрим их на примере ввода функции МИН в поле Лог_выражение при выполнении задания примера 2 (см. рис. 3):

- установить курсор в поле Лог_выражение;
- из выпадающего списка в поле адреса ячейки (слева от строки формул) выбрать нужную функцию (если она там отсутствует, то воспользоваться опцией *Другие функции...*) – в нашем случае это функция МИН;
- в открывшемся диалоговом окне функции МИН в поле *Число1* указать диапазон проверяемых ячеек D3:G3 (для этого достаточно выделить указанный диапазон мышью);
- для возврата в диалоговое окно функции ЕСЛИ щелкнуть на имени функции ЕСЛИ в строке формул и продолжить ввод данных согласно изложенному в примере 2 порядку.

Такой способ ввода наиболее удобен при создании сложных формул, содержащих несколько функций с большим количеством аргументов.

¹ Microsoft Excel 2010 допускает до 64 уровней вложенности функций.

Если необходимо отредактировать формулу с вложенными функциями, то для вызова диалогового окна любой из входящих в нее функций нужно в режиме редактирования формулы (непосредственно в ячейке или строке формул) установить курсор на имя функции и нажать кнопку .

Пример применения вложенных функций ЕСЛИ рассмотрен в подразд. 3.7.

3.4. Функция СЧЁТЕСЛИ

Функция СЧЁТЕСЛИ подсчитывает количество непустых ячеек в указанном диапазоне, удовлетворяющих заданному условию. В библиотеке встроенных функций Excel она находится в категории *Статистические*.

Общая форма записи функции СЧЁТЕСЛИ:

СЧЁТЕСЛИ(Диапазон; Критерий),

где **Диапазон** – диапазон ячеек, в котором подсчитывается количество непустых ячеек;

Критерий – условие в форме числа, выражения или текста, которое определяет, какие ячейки надо подсчитывать.

Пример 3. В ячейках D10:G10 (см. рис. 1) подсчитать общее количество пятерок по каждой учебной дисциплине.

Порядок действий:

1. Установить курсор в первую результирующую ячейку D10.
2. Запустить Мастер функций, выбрать функцию СЧЁТЕСЛИ и в открывшемся диалоговом окне этой функции ввести ее аргументы (рис. 4):

в поле **Диапазон** – диапазон ячеек в графе «Математика», в которых будет проверяться выполнение заданного условия (D3:D9);

в поле **Критерий** – проверяемое условие (оценка 5).

3. Нажать кнопку ОК, вследствие чего в ячейке D10 автоматически сформируется формула =СЧЁТЕСЛИ(D3:D9; "=5"), и Excel отобразит в ней результат вычислений – число 4 (количество пятерок по математике).

4. Скопировать формулу из ячейки D10 в ячейки E10:G10 с помощью маркера заполнения.

Результат выполнения примера 3 в ячейках D10:G10 представлен в подразд. 3.6 (стр. 15).




Рис. 4. Диалоговое окно функции СЧЁТЕСЛИ

3.5. Функция СЧЁТЕСЛИМН

В тех случаях, когда для решения задачи нужно выполнить подсчет количества значений, удовлетворяющих не одному, а сразу нескольким условиям, применяется функция СЧЁТЕСЛИМН, в диалоговом окне которой можно перечислить до 127 пар проверяемых диапазонов и соответствующих им условий. Например, если в примере 2 дополнительно понадобится подсчитать количество пятерок по математике в группе 476 (см. рис. 1) и вывести результат в ячейке D11, то нужно выполнить следующие действия:

- 1) установить курсор в результирующую ячейку D11;
- 2) запустить Мастер функций, выбрать функцию СЧЁТЕСЛИМН и в открывшемся диалоговом окне этой функции ввести ее аргументы (рис. 5):
 - в поле Диапазон_условия1 – диапазон ячеек в графе «Математика», в которых будет проверяться выполнение первого условия (D3:D9);
 - в поле Условие1 – первое проверяемое условие (оценка 5);
 - в поле Диапазон_условия2 – диапазон ячеек в графе «Группа», в которых будет проверяться выполнение второго условия (C3:C9);
 - в поле Условие2 – второе проверяемое условие (476);
- 3) нажать кнопку ОК, вследствие чего в ячейке D11 автоматически сформируется формула =СЧЁТЕСЛИМН(D3:D9; 5; C3:C9; "476"), и Excel отобразит в ней результат вычислений – число 3.




Рис. 5. Диалоговое окно функции СЧЁТЕСЛИМН

3.6. Функция СУММЕСЛИ

Функция СУММЕСЛИ суммирует ячейки, заданные указанным условием. В библиотеке встроенных функций Excel она находится в категории *Математические*.

Общая форма записи функции СУММЕСЛИ:

СУММЕСЛИ(Диапазон; Критерий; Диапазон_суммирования),

где **Диапазон** – диапазон проверяемых ячеек;

Критерий – условие в форме числа, выражения или текста, определяющее суммируемые ячейки;

Диапазон_суммирования – фактические ячейки для суммирования. Если диапазон суммирования не указан, то будут суммироваться ячейки, заданные аргументом **Диапазон**.

Пример 4. В ячейках J11 и J12 (см. рис. 1) вычислить суммы стипендии, назначенной студентам групп 47а и 47б.

Порядок действий:

1. Установить курсор в ячейку J11.
2. Запустить Мастер функций, выбрать функцию СУММЕСЛИ и в открывшемся диалоговом окне этой функции ввести ее аргументы (рис. 6):

в поле **Диапазон** – диапазон ячеек в графе «Группа», в которых будет проверяться выполнение заданного условия (С3:С9);

в поле **Критерий** – ячейку I11, которая содержит проверяемое условие (47а);

в поле **Диапазон_суммирования** – диапазон ячеек в графе «Стипендия, руб.», значения которых будут суммироваться в случае выполнения проверяемого условия (J3:J9).




Рис. 6. Диалоговое окно функции СУММЕСЛИ

3. Нажать кнопку ОК, вследствие чего в ячейке J11 автоматически сформируется формула =СУММЕСЛИ(С3:С9; I11; J3:J9), и Excel отобразит в ней результат вычислений – число 5000 (общую сумму стипендии студентов группы 47а).

Вычисление общей суммы стипендии студентов группы 47б в ячейке J12 можно выполнить аналогично, но более эффективно применить для этого автозаполнение. С этой целью в формуле в ячейке J11 нужно поменять относительные адреса ячеек проверяемого и суммируемого диапазонов на абсолютные (для этого можно воспользоваться клавишей F4). В результате формула в ячейке J11 будет иметь вид =СУММЕСЛИ(\$C\$3:\$C\$9; I11; \$J\$3:\$J\$9), после чего ее нужно скопировать из ячейки J11 в ячейку J12 с помощью маркера заполнения.

Итоговая таблица результатов сессии (см. рис. 1) после выполнения примеров 1 – 4 представлена на рис. 7.

	A	B	C	D	E	F	G	H	I	J
1	№ п/п	Фамилия И.О. студента	Группа	Оценки по дисциплинам					Список студентов (ср. балл ≥ 4)	Стипендия, руб.
2				Математика	Физика	Информатика	История	Средний балл		
3	1	Авдеев К.В.	47а	4	5	3	3	=СРЗНАЧ(D3:G3)	=ЕСЛИ(H3>=4;B3;"")	=ЕСЛИ(МИН(D3:G3)>=4; 5000; 0)
4	2	Белова А.С.	47б	4	4	5	5	=СРЗНАЧ(D4:G4)	=ЕСЛИ(H4>=4;B4;"")	=ЕСЛИ(МИН(D4:G4)>=4; 5000; 0)
5	3	Ерофеев Д.И.	47а	5	5	4	5	=СРЗНАЧ(D5:G5)	=ЕСЛИ(H5>=4;B5;"")	=ЕСЛИ(МИН(D5:G5)>=4; 5000; 0)
6	4	Иванов И.П.	47б	5	5	5	5	=СРЗНАЧ(D6:G6)	=ЕСЛИ(H6>=4;B6;"")	=ЕСЛИ(МИН(D6:G6)>=4; 5000; 0)
7	5	Котова Н.В.	47б	5	4	3	5	=СРЗНАЧ(D7:G7)	=ЕСЛИ(H7>=4;B7;"")	=ЕСЛИ(МИН(D7:G7)>=4; 5000; 0)
8	6	Морозов Ф.Д.	47а	3	3	3	3	=СРЗНАЧ(D8:G8)	=ЕСЛИ(H8>=4;B8;"")	=ЕСЛИ(МИН(D8:G8)>=4; 5000; 0)
9	7	Федюнин И.Н.	47б	5	4	4	5	=СРЗНАЧ(D9:G9)	=ЕСЛИ(H9>=4;B9;"")	=ЕСЛИ(МИН(D9:G9)>=4; 5000; 0)
10	Количество 5 по дисциплинам			=СЧЁТЕСЛИ(D3:D9;"=5")	=СЧЁТЕСЛИ(E3:E9;"=5")	=СЧЁТЕСЛИ(F3:F9;"=5")	=СЧЁТЕСЛИ(G3:G9;"=5")			
11								Стипендия по группам, руб.	47а	=СУММЕСЛИ(\$C\$3:\$C\$9; I11; \$J\$3:\$J\$9)
12									47б	=СУММЕСЛИ(\$C\$3:\$C\$9; I12; \$J\$3:\$J\$9)

а

	A	B	C	D	E	F	G	H	I	J
1	№ п/п	Фамилия И.О. студента	Группа	Оценки по дисциплинам					Список студентов (ср. балл ≥ 4)	Стипендия, руб.
2				Математика	Физика	Информатика	История	Средний балл		
3	1	Авдеев К.В.	47а	4	5	3	3	3,75		0
4	2	Белова А.С.	47б	4	4	5	5	4,50	Белова А.С.	5000
5	3	Ерофеев Д.И.	47а	5	5	4	5	4,75	Ерофеев Д.И.	5000
6	4	Иванов И.П.	47б	5	5	5	5	5,00	Иванов И.П.	5000
7	5	Котова Н.В.	47б	5	4	3	5	4,25	Котова Н.В.	0
8	6	Морозов Ф.Д.	47а	3	3	3	3	3,00		0
9	7	Федюнин И.Н.	47б	5	4	4	5	4,50	Федюнин И.Н.	5000
10	Количество 5 по дисциплинам			4	3	2	5			
11								Стипендия по группам, руб.	47а	5000
12									47б	15000

б

Рис. 7. Итоговая таблица результатов сессии
в режимах отображения формул (а) и отображения значений (б)

3.7. Организация разветвлений с помощью вложенных функций ЕСЛИ

С помощью вложенных функций ЕСЛИ можно организовать вычислительный процесс с множеством разветвлений. Рассмотрим их применение на конкретном примере.

Пример 5. Вычислить значение функции y при изменении аргумента x в диапазоне от 5 до 15 с шагом 1:

$$y = \begin{cases} 2,5x & \text{при } x < 7; \\ \sqrt[3]{x^2} & \text{при } 7 \leq x \leq 11; \\ x + 0,5 & \text{при } x > 11. \end{cases} \quad (1)$$

Фрагмент рабочего листа Excel с результатами расчета представлен на рис. 8, примеры записи математических выражений в Excel подробно рассмотрены в работе [4, с. 14].




Рис. 8. Пример вложенных функций ЕСЛИ

Порядок действий:

1. В ячейки A3:A13 с помощью автозаполнения с арифметической прогрессией ввести значения x от 5 до 15 с шагом 1.

2. Установить курсор в первую расчетную ячейку В3, запустить Мастер функций и выбрать функцию ЕСЛИ.

3. В открывшемся диалоговом окне (рис. 9) ввести:

в поле **Лог_выражение** – первое условие из формулы (1). Для определения значения y в ячейке В3 исходное значение x находится в ячейке А3, поэтому условие будет иметь вид $A3 < 7$;

в поле **Значение_если_истина** – соответствующее первому условию формулы (1) выражение для расчета значения y ($2,5 * A3$);

поле **Значение_если_ложь** должно содержать выражение для расчета значения y , если условие $x < 7$ не выполняется (т. е. при $x \geq 7$), но в формуле (1) для такого случая приведены два разных выражения: второе – при $x \leq 11$ и третье – при $x > 11$. Для выбора нужного варианта необходимо в поле **Значение_если_ложь** опять вставить функцию ЕСЛИ, для чего щелкнуть мышью в поле адреса ячейки на имени функции ЕСЛИ.




Рис. 9. Ввод первой (внешней) функции ЕСЛИ

4. В открывшемся новом диалоговом окне функции ЕСЛИ (с пустыми полями аргументов) ввести второе проверяемое условие из формулы (1) и соответствующие выражения, как показано на рис. 10.

5. Нажать кнопку ОК, вследствие чего в ячейке В3 автоматически сформируется формула, приведенная на рис. 8, и Excel отобразит в ней результат расчета – число 12,5.

6. Скопировать введенную в ячейку В3 формулу в ячейки В4:В13 с помощью маркера заполнения.




Рис. 10. Ввод второй (вложенной) функции ЕСЛИ

Поскольку каждая формула в ячейках В3:В13 содержит в себе все три заданных выражения для расчета y , они будут выдавать правильные результаты для любых произвольных значений x , введенных в соответствующие ячейки А3:А13, в том числе из разных диапазонов числовой оси. Например, если в ячейку А3 ввести число 10, то в ячейке В3 будет результат 4,6 (расчет по второму выражению), если в ячейку А3 ввести число 15 – в ячейке В3 будет результат 15,5 (расчет по третьему выражению).

3.8. Логические функции НЕ, И, ИЛИ

Для расширения функционала проверки выполнения условий в Excel имеются функции НЕ, И, ИЛИ. В библиотеке встроенных функций они находятся в категории *Логические* и во многих случаях применяются совместно с функцией ЕСЛИ.

Аргументами и результатами этих функций являются логические значения *Истина* или *Ложь*, а их действие аналогично соответствующим операциям алгебры логики:

функция НЕ(Логическое_значение) принимает в виде аргумента всего одно логическое значение и меняет его на противоположное (рис. 11, а), т. е. значение *Истина* она изменит на *Ложь* и наоборот;

функция И(Логическое_значение1; [Логическое_значение2]; ...) возвращает логическое значение *Истина*, если все аргументы функции вернули истинное значение. Если хотя бы один аргумент возвращает значение *Ложь*, то функция также вернет значение *Ложь* (рис. 11, б);

функция ИЛИ(Логическое_значение1; [Логическое_значение2]; ...) возвращает логическое значение *Истина*, если хотя бы один аргумент функции вернет истинное значение (рис. 11, в).

C2 fx =НЕ(A2)			
	A	B	C
1	a		¬a
2	0		ИСТИНА
3	1		ЛОЖЬ

C2 fx =И(A2; B2)			
	A	B	C
1	a	b	a ∧ b
2	0	0	ЛОЖЬ
3	0	1	ЛОЖЬ
4	1	0	ЛОЖЬ
5	1	1	ИСТИНА

C2 fx =ИЛИ(A2; B2)				
	A	B	C	D
1	a	b	a ∨ b	
2	0	0	ЛОЖЬ	
3	0	1	ИСТИНА	
4	1	0	ИСТИНА	
5	1	1	ИСТИНА	

а
б
в

Рис. 11. Результаты логических функций НЕ (а), И (б), ИЛИ (в)

Рассмотрим применение логических функций на конкретном примере.

Пример 6. Составить таблицу истинности логического выражения $\bar{b} \vee a \wedge b$, полученный результат представить с помощью констант 1 (*Истина*) и 0 (*Ложь*).

Решим поставленную задачу двумя способами.

Способ 1. Построим результирующую таблицу истинности, выполняя каждую логическую операцию по отдельности (рис. 12):

1) в ячейках A2:B5 введем все возможные комбинации значений логических переменных *a* и *b*;

2) в ячейках C2, D2, E2 введем по очереди логические функции заданного выражения в соответствии с приоритетом их выполнения;

3) в ячейке F2 введем функцию ЕСЛИ для преобразования логических значений *Истина* и *Ложь* соответственно в числовые значения 1 и 0;

4) скопируем формулу из ячейки F2 в ячейки F3:F5 с помощью маркера заполнения.

	A	B	C	D	E	F
1	a	b	$\neg b$	$a \wedge b$	$\neg b \vee a \wedge b$	
2	0	0	=НЕ(B2)	=И(A2; B2)	=ИЛИ(C2; D2)	=ЕСЛИ(E2=ИСТИНА; 1; 0)
3	0	1	=НЕ(B3)	=И(A3; B3)	=ИЛИ(C3; D3)	=ЕСЛИ(E3=ИСТИНА; 1; 0)
4	1	0	=НЕ(B4)	=И(A4; B4)	=ИЛИ(C4; D4)	=ЕСЛИ(E4=ИСТИНА; 1; 0)
5	1	1	=НЕ(B5)	=И(A5; B5)	=ИЛИ(C5; D5)	=ЕСЛИ(E5=ИСТИНА; 1; 0)

а

	A	B	C	D	E	F
1	a	b	$\neg b$	$a \wedge b$	$\neg b \vee a \wedge b$	
2	0	0	ИСТИНА	ЛОЖЬ	ИСТИНА	1
3	0	1	ЛОЖЬ	ЛОЖЬ	ЛОЖЬ	0
4	1	0	ИСТИНА	ЛОЖЬ	ИСТИНА	1
5	1	1	ЛОЖЬ	ИСТИНА	ИСТИНА	1

б

Рис. 12. Результат выполнения примера 6 (способ 1) в режимах отображения формул (а) и отображения значений (б)

В функции ЕСЛИ аргумент Лог_выражение (см. подразд. 3.2) может принимать только значения *Истина* или *Ложь*, поэтому в формулах в ячейках F2:F5 можно опустить оператор сравнения (=ИСТИНА) и записать их более компактно. Например, в ячейке F2 формула может иметь вид =ЕСЛИ(E2; 1; 0).

Способ 2. Построим результирующую таблицу истинности, объединив все логические операции в одной формуле с помощью вложенных функций (рис. 13):

- 1) в ячейках A2:B5 введем все возможные комбинации значений логических переменных *a* и *b*;
- 2) в ячейке C2 введем функцию ЕСЛИ с вложенными функциями ИЛИ, НЕ, И;

3) скопируем формулу из ячейки F2 в ячейки F3:F5 с помощью маркера заполнения.

	A	B	C
1	a	b	$\neg b \vee a \wedge b$
2	0	0	=ЕСЛИ(ИЛИ(НЕ(B2);И(A2;B2));1;0)
3	0	1	=ЕСЛИ(ИЛИ(НЕ(B3);И(A3;B3));1;0)
4	1	0	=ЕСЛИ(ИЛИ(НЕ(B4);И(A4;B4));1;0)
5	1	1	=ЕСЛИ(ИЛИ(НЕ(B5);И(A5;B5));1;0)

Рис. 13. Вложенные логические функции при выполнении примера 6 в режиме отображения формул

3.9. Контрольные вопросы

- 1) Какая функция Excel используется для подсчета количества ячеек, значения которых удовлетворяют некоторому условию?
- 2) В чем заключается различие ввода проверяемого условия в функциях СЧЁТЕСЛИ и ЕСЛИ?
- 3) Какие аргументы имеет функция СУММЕСЛИ?
- 4) Как можно отредактировать формулу с вложенными функциями?
- 5) Могут ли быть вложенными функции ЕСЛИ?
- 6) Приведите примеры использования функции ЕСЛИ и логических функций НЕ, И, ИЛИ.

3.10. Задания

Задание 1. Вычисления с проверкой условий.

1. Откройте Excel-шаблон.
2. Сохраните Excel-шаблон с именем Фамилия_Excel-3 (здесь и далее вместо слова «Фамилия» укажите вашу фамилию) в личной папке. Все дальнейшие действия выполняйте в этом файле.
3. В свойствах файла в поле *Примечания* укажите свои фамилию и группу.
4. Создайте новый рабочий лист, назвав его «Лаб3_31».
5. В верхнем колонтитуле справа введите свой учебный шифр, в нижнем колонтитуле справа – фамилию и инициалы преподавателя.

6. Создайте таблицу с результатами сессии по образцу рис. 14. При этом:
в ячейке «Фамилия И.О.» создайте примечание, в котором укажите дату выполнения работы;

в столбце «№ п/п» с помощью автозаполнения введите порядковые номера строк от 1 до K , где K – количество студентов в вашей подгруппе;

в строке с вашим порядковым номером в столбце «№ п/п» исправьте номер на свой учебный шифр и в соответствующей ячейке столбца «Фамилия И.О.» введите свои фамилию и инициалы;

в остальных строках в столбце «Фамилия И.О.» в произвольном порядке введите фамилии всех студентов вашей подгруппы;

во всех ячейках столбца «Группа» укажите свою группу;

ячейки в столбцах оценок по дисциплинам заполните произвольным образом числами 2, 3, 4, 5 (в каждой строке должно быть не менее двух оценок).

	A	B	C	D	E	F	G	H	I	J	K	L
1	Результаты сессии											
2	№ п/п	Фамилия И.О.	Группа	Оценки				Средний балл (СрБ)	Всего баллов	Фамилия И.О. (СрБ ≥ 3)	Отличники (+)	Перевод на след. курс (да/нет)
3				Математика	Физика	Информатика	Химия					
4	1	3		4	2					
5	2	5	5		5					
6									
7	Средний балл по дисциплинам									Всего		
8												
9	Итоги											
10	Всего оценок											
11	в том числе:											
12	5											
13	4											
14	3											
15	2											

Рис. 14. Образец таблицы для выполнения задания 1

7. Значения в ячейках, выделенных заливкой, вычислите с применением встроенных функций Excel:

в ячейках строки «Средний балл по дисциплинам» с помощью функции СРЗНАЧ определите средний балл по каждой дисциплине (с одним десятичным знаком);

в ячейках столбца «Средний балл (СрБ)» с помощью функции СРЗНАЧ рассчитайте средний балл по каждому студенту (с одним десятичным знаком);

в ячейках столбца «Всего баллов» с помощью функции СУММ вычислите общую сумму баллов по каждому студенту;

в ячейках столбца «Фамилия И.О. (СрБ ≥ 3)» с помощью функции ЕСЛИ выведите фамилии студентов со средним баллом не ниже 3;

в ячейках столбца «Отличники (+)» с помощью функции ЕСЛИ выведите признак «+», если студент имеет средний балл по всем сданным экзаменам 5, в противном случае оставьте ячейку пустой. В последней ячейке этого столбца выведите количество отличников;

в ячейках столбца «Перевод на след. курс (да/нет)» с помощью функции ЕСЛИ выведите сообщение «да», если студент переведен на следующий курс (обязательным условием для перевода студента является отсутствие у него двоек по дисциплинам), иначе выведите сообщение «нет». В последней ячейке этого столбца выведите количество студентов, переведенных на следующий курс;

в таблице «Итоги» с помощью функций СЧЁТ и СЧЁТЕСЛИ вычислите общее количество оценок и количество оценок каждого вида.

8. Сохраните содержимое рабочей книги.

За д а н и е 2. Суммирование с проверкой условия.

1. Откройте свою рабочую книгу Фамилия_Excel-3, создайте в ней новый рабочий лист с именем «Лаб3-з2», на котором выполните приведенные ниже задания.

2. Создайте таблицу с результатами тестирования по образцу рис. 15. Ячейки с наименованиями тестов и ячейки с фамилиями заполните с помощью автозаполнения.

3. В ячейках C8:E16 столбца «Коэффициент выполнения теста» установите проверку данных, задав условие, что вводимое значение должно быть в диапазоне [0; 1]. Предусмотрите вывод соответствующего сообщения при ошибке (в заголовке сообщения об ошибке укажите свою фамилию).

4. Введите в ячейки C8:E16 столбца «Коэффициент выполнения теста» произвольные числа в диапазоне [0; 1].

5. Значения в ячейках с заливкой вычислите с применением встроенных функций Excel (для удобства копирования формул используйте в них относительные, абсолютные и смешанные адреса ячеек):

в ячейках F8:H16 столбца «Количество баллов» рассчитайте оценку в баллах, полученную каждым студентом за отдельный тест, как произведение коэффициента выполнения теста на максимальное количество баллов за этот тест;

в ячейках I8:I16 столбца «Количество баллов» (в графе «Всего») вычислите для каждого студента суммарный балл за все тесты;

в столбце «СУММПРОИЗВ» для каждого студента определите суммарный балл за все тесты другим способом – с помощью функции СУММПРОИЗВ (без использования результатов расчетов из столбца «Количество баллов»).

Сравните результаты в столбцах I и J и убедитесь в их идентичности;

в столбце «Количество студентов» примените функцию СЧЁТЕСЛИ;

в ячейках столбца «Сумма баллов» примените функцию СУММЕСЛИ;

в столбце «Средний балл на одного студента» вычислите средний балл по всем тестам в расчете на одного студента в каждой группе (учтите, что функцию СРЗНАЧ в этом случае применять нельзя).

6. Сохраните содержимое рабочей книги.

	A	B	C	D	E	F	G	H	I	J
1	Критерии оценивания									
2	Наименование теста	Тест1	Тест2	Тест3						
3	Максимальное количество баллов	25	30	45						
4										
5	Результаты тестирования									
6	Группа	Фамилия	Коэффициент выполнения теста			Количество баллов				СУММ ПРОИЗВ
7			Тест1	Тест2	Тест3	Тест1	Тест2	Тест3	Всего	
8	47а	Фамилия1	0,56	0,94	0,36					
9	47а	Фамилия2					
10	47б	Фамилия3					
11	47в	Фамилия4					
12	47б	Фамилия5					
13	47в	Фамилия6					
14	47б	Фамилия7					
15	47а	Фамилия8					
16	47б	Фамилия9					
17										
18	Итоги по группам									
19	Группа	Количество студентов	Сумма баллов			Средний балл на одного студента				
20			Тест1	Тест2	Тест3					
21	47а									
22	47б									
23	47в									

Рис. 15. Образец таблицы для выполнения задания 2

Задание 3. Вложенные функции ЕСЛИ.

1. Откройте свою рабочую книгу Фамилия_Excel-3, создайте в ней новый рабочий лист с именем «Лаб3-з3», на котором выполните приведенные ниже задания.

2. В соответствии с индивидуальным вариантом из табл. 1 создайте формулу для вычисления значения функции с применением вложенных функций ЕСЛИ аналогично примеру 5. При необходимости задайте значение $\pi = 3,1415926$. Образец выполнения задания для варианта 19 представлен на рис. 16.

3. Скопируйте созданную формулу в две соседние ячейки ниже и проверьте результат расчета при всех контрольных значениях x , указанных в последней графе табл. 1.

4. Сохраните содержимое рабочей книги.

D8		fx		=ЕСЛИ(C8<0,5;SIN(6*C8)^3;ЕСЛИ(C8<=1,5;EXP(\$A\$8+C8)-4;COS((C8+7)^2)))				
	A	B	C	D	E	F	G	H
1	19	$y = \begin{cases} \sin^3 6x & x < 0,5 \\ e^{w+x} - 4 & 0,5 \leq x \leq 1,5 \\ \cos(x+7)^2 & x > 1,5 \end{cases}$			w = 0,65	При x = 0,4 y = 0,31 При x = 1,5 y = 4,58 При x = 2,6 y = -0,49		
2								
3								
4								
5								
6								
7	w		x	y				
8	0,65		0,4	0,31				
9			1,5	4,58				
10			2,6	-0,49				

Рис. 16. Образец выполнения задания 3

Таблица 1

Функции для вычисления с помощью вложенных функций ЕСЛИ

Вариант	Функция	Исходные данные	Контрольные значения
1	2	3	4
0	$y = \begin{cases} \ln ax^2 + b - 0,5 & x < 0,2 \\ e^{ax} - ab & 0,2 \leq x \leq 2,2 \\ \sin^2(x+a) & x > 2,2 \end{cases}$	$a = 0,12;$ $b = -4,4$	При x = 0 y = 0,98 При x = 2,2 y = 1,83 При x = 4,4 y = 0,96

Продолжение табл. 1

1	2	3	4
1	$y = \begin{cases} \lg xk & x < -1 \\ \cos(x+k)^3 & -1 \leq x \leq 0 \\ \sqrt[3]{x} + e^{kx} & x > 0 \end{cases}$	$k = 0,04$	При $x = -1,1$ $y = -1,36$ При $x = 0$ $y = 1$ При $x = 1,1$ $y = 2,08$
2	$y = \begin{cases} \operatorname{tg}^2(x+\pi) & x < 0 \\ d + e^{-x\pi} & 0 \leq x \leq 1 \\ \ln \sqrt{dx} & x > 1 \end{cases}$	$d = 6,2$	При $x = -0,1$ $y = 0,01$ При $x = 1$ $y = 6,24$ При $x = 2,1$ $y = 1,28$
3	$y = \begin{cases} x + \lg(a+b) & x < 0,3 \\ \sin \sqrt{7x+b} & 0,3 \leq x \leq 0,7 \\ x^2 + \frac{e^x}{b+x} & x > 0,7 \end{cases}$	$a = 98,3;$ $b = 4,5$	При $x = 0,26$ $y = 2,27$ При $x = 0,7$ $y = 0,08$ При $x = 1,14$ $y = 1,85$
4	$y = \begin{cases} 0,3w - e^{x-k} & x < 2 \\ \cos^3(wx+k) & 2 \leq x \leq 7 \\ \ln(x-0,3) & x > 7 \end{cases}$	$w = 2,81;$ $k = 0,95$	При $x = 1,5$ $y = -0,89$ При $x = 7$ $y = -0,01$ При $x = 12,5$ $y = 2,5$
5	$y = \begin{cases} e^{2x} - \sqrt[3]{d^4} & x < 1 \\ \lg(0,5d + x^2) & 1 \leq x \leq 4 \\ \sin(x+d^2) & x > 4 \end{cases}$	$d = 4,4$	При $x = 0,7$ $y = -3,15$ При $x = 4$ $y = 1,26$ При $x = 7,3$ $y = 1$
6	$y = \begin{cases} \ln x+k & x < -2 \\ k + 0,65e^{x-k} & -2 \leq x \leq 2 \\ \sqrt[3]{kx} + \cos^2 kx & x > 2 \end{cases}$	$k = 2,1$	При $x = -2,4$ $y = -1,2$ При $x = 2$ $y = 2,69$ При $x = 6,4$ $y = 2,79$
7	$y = \begin{cases} a + \sin^2 x & x < 0 \\ \sqrt{\lg(a-x)} & 0 \leq x \leq 6 \\ e^{\sqrt{x-b}} & x > 6 \end{cases}$	$a = 7,13;$ $b = 0,91$	При $x = -0,6$ $y = 7,45$ При $x = 6$ $y = 0,23$ При $x = 12,6$ $y = 30,54$

Продолжение табл. 1

1	2	3	4
8	$y = \begin{cases} e^{wx} + \sqrt{w} & x < 1,5 \\ \sin\left(x^2 - \frac{\pi}{2}\right) & 1,5 \leq x \leq 6,5 \\ \ln 0,39x^2 & x > 6,5 \end{cases}$	$w = 1,57$	При $x = 1$ $y = 6,06$ При $x = 6,5$ $y = 0,16$ При $x = 12$ $y = 4,03$
9	$y = \begin{cases} d \sin^2 x & x < 0 \\ e^{0,2d\sqrt{x}} & 0 \leq x \leq 5 \\ \lg x^2 - 0,2x & x > 5 \end{cases}$	$d = 2,5$	При $x = -0,5$ $y = 0,57$ При $x = 5$ $y = 3,06$ При $x = 10,5$ $y = -0,06$
10	$y = \begin{cases} b \sin x & x < 3,5 \\ \ln^2 x - e^{-bx} & 3,5 \leq x \leq 8,5 \\ \cos \sqrt{(x+b)^3} & x > 8,5 \end{cases}$	$b = 0,91$	При $x = 3$ $y = 0,13$ При $x = 8,5$ $y = 4,58$ При $x = 14$ $y = 0,52$
11	$y = \begin{cases} e^{\cos x } & x < -2 \\ \lg \sqrt{(x+a)^3} & -2 \leq x \leq 5 \\ a\sqrt{ \sin x } & x > 5 \end{cases}$	$a = 3,8$	При $x = -2,7$ $y = 0,4$ При $x = 5$ $y = 1,42$ При $x = 12,7$ $y = 1,39$
12	$y = \begin{cases} 2 \cos\left(x^2 - \frac{\pi}{2}\right) & x < 0,5 \\ 2x - 3 \ln x - 3 & 0,5 \leq x \leq 0,6 \\ e^{-kx} - \frac{x^2}{3 + 0,2x^2} & x > 0,6 \end{cases}$	$k = 0,1$	При $x = 0,49$ $y = 0,48$ При $x = 0,6$ $y = -0,27$ При $x = 0,71$ $y = 0,77$
13	$y = \begin{cases} e^{bx} - \sqrt{2x} & x < 2 \\ \lg x^2 - x + 1,8 & 2 \leq x \leq 3 \\ \sin^2(x^3 + 1) & x > 3 \end{cases}$	$b = 0,7$	При $x = 1,9$ $y = 1,83$ При $x = 3$ $y = -0,25$ При $x = 4,1$ $y = 0,52$
14	$y = \begin{cases} \cos(x^2 + a) & x < 0 \\ \ln^2(1+x) - 1,5 & 0 \leq x \leq 1 \\ e^{-ax} + 0,5 & x > 1 \end{cases}$	$a = 2$	При $x = -0,1$ $y = -0,43$ При $x = 1$ $y = -1,02$ При $x = 2,1$ $y = 0,51$

1	2	3	4
15	$y = \begin{cases} e^{-1,57x} - 7 & x < 1,2 \\ x + 3\sqrt{\sin\left(\frac{1}{x}\right)} & 1,2 \leq x \leq 3,2 \\ d + \lg \sqrt{x+7} & x > 3,2 \end{cases}$	$d = 4,7$	При $x = 1$ $y = -6,79$ При $x = 3,2$ $y = 3,87$ При $x = 5,4$ $y = 5,25$
16	$y = \begin{cases} \ln^2 \sqrt{x^2 - 0,5} & x < 2 \\ \cos\left(\frac{x+a}{x-b}\right) & 2 \leq x \leq 3 \\ e^{ax-b} + 0,7 & x > 3 \end{cases}$	$a = 0,5;$ $b = 1$	При $x = 1,9$ $y = 0,32$ При $x = 3$ $y = -0,18$ При $x = 4,1$ $y = 3,56$
17	$y = \begin{cases} e^{x+k} - 2 & x < 0 \\ \sin^3(x + \sqrt{k}) & 0 \leq x \leq 1 \\ \lg(x + 7\sqrt{x}) - x & x > 1 \end{cases}$	$k = 0,5$	При $x = -0,1$ $y = -0,51$ При $x = 1$ $y = 0,97$ При $x = 2,1$ $y = -1,01$
18	$y = \begin{cases} \ln^2 ax - 2 & x < 4,5 \\ 5 - e^{-x+\sqrt{a}} & 4,5 \leq x \leq 6,5 \\ a + \cos \sqrt{x+a^2} & x > 6,5 \end{cases}$	$a = 3,4$	При $x = 4,3$ $y = 5,2$ При $x = 6,5$ $y = 4,99$ При $x = 8,7$ $y = 3,19$

Задание 4. Работа с логическими функциями.

1. Откройте свою рабочую книгу Фамилия_Excel-3, создайте в ней новый рабочий лист с именем «Лаб3-34», на котором выполните приведенные ниже задания.

2. В соответствии с индивидуальным вариантом из табл. 2 составьте таблицу истинности заданного логического выражения двумя способами аналогично примеру 6. Оформите работу по образцу рис. 12 и 13. Убедитесь в том, что полученные разными способами результаты являются идентичными.

3. Сохраните содержимое рабочей книги.

Логические выражения

Вариант	Выражение	Вариант	Выражение	Вариант	Выражение	Вариант	Выражение
0	$a \wedge \overline{b \vee c}$	5	$a \wedge (\overline{b \vee c})$	10	$\overline{a \wedge b \wedge c}$	15	$\overline{a \vee b \vee c}$
1	$\overline{a} \wedge b \vee c$	6	$a \wedge \overline{b} \wedge \overline{c}$	11	$\overline{a} \wedge (b \vee c)$	16	$\overline{a} \vee \overline{b} \wedge c$
2	$a \vee \overline{b} \wedge c$	7	$a \wedge (\overline{b \vee c})$	12	$\overline{a} \vee b \wedge \overline{c}$	17	$\overline{a \vee b} \wedge \overline{c}$
3	$a \vee \overline{b} \wedge \overline{c}$	8	$\overline{a \wedge b \vee c}$	13	$\overline{a} \wedge \overline{b \vee c}$	18	$\overline{a} \wedge \overline{b} \wedge \overline{c}$
4	$\overline{a \vee b} \wedge c$	9	$\overline{a \wedge b \vee c}$	14	$\overline{a} \wedge \overline{b \vee c}$		

Библиографический список

1. Гаврилов М. В. Информатика и информационные технологии: Учебник / М. В. Гаврилов, В. А. Климов. М.: Юрайт, 2016. 383 с.
2. Патеюк А. Г. Обработка данных в табличном процессоре Microsoft Excel 2010. Часть 1. Основы вычислений в электронных таблицах: Учебное пособие / А. Г. Патеюк / Омский гос. ун-т путей сообщения. Омск, 2014. 58 с.
3. Настащук Н. А. Обработка данных в табличном процессоре Microsoft Excel 2010. Часть 2. Работа с диаграммами и списками в электронных таблицах: Учебное пособие / Н. А. Настащук / Омский гос. ун-т путей сообщения. Омск, 2014. 87 с.
4. Сидорова Е. А. Лабораторный практикум по работе в табличном процессоре Microsoft Excel 2010. Часть 2. Создание формул. Встроенные функции: Практикум / Е. А. Сидорова, С. П. Железняк / Омский гос. ун-т путей сообщения. Омск, 2018. 37 с.

Учебное издание

СИДОРОВА Елена Анатольевна,
ЖЕЛЕЗНЯК Светлана Петровна

ЛАБОРАТОРНЫЙ ПРАКТИКУМ ПО РАБОТЕ
В ТАБЛИЧНОМ ПРОЦЕССОРЕ MICROSOFT EXCEL 2010

Часть 3

ВЫЧИСЛЕНИЯ С ПРОВЕРКОЙ УСЛОВИЙ

Практикум

Редактор Н. А. Майорова

Подписано в печать 08.06.2018. Формат 60 × 84 ¹/₁₆.
Офсетная печать. Бумага офсетная. Усл. печ. л. 1,9. Уч.-изд. л. 2,1.
Тираж 500 экз. Заказ .

**

Редакционно-издательский отдел ОмГУПСа
Типография ОмГУПСа

*

644046, г. Омск, пр. Маркса, 35